

```

;;; olirelay.asm

        processor 16f1847
        include p16f1847.inc

#ifdef  __DEBUG
        __CONFIG __CONFIG1,_FOSC_INTOSC & _WDTE_OFF & _PWRTE_OFF & _MCLRE_ON & _CP_OFF
        & _CPD_OFF & _BOREN_ON & _CLKOUTEN_ON & _IESO_ON & _FCMEN_ON
#else
        __CONFIG __CONFIG1,_FOSC_HS & _WDTE_ON & _PWRTE_OFF & _MCLRE_ON & _CP_OFF &
        _CPD_OFF & _BOREN_ON & _CLKOUTEN_OFF & _IESO_ON & _FCMEN_ON
#endif
        __CONFIG __CONFIG2,_WRT_ALL & _PLLEN_OFF & _STVREN_ON & _BORV_LO & _LVP_ON

;;; example program to control the Olimex PIC-IO relay/optoisolator board loaded
;;; with a PIC16f1847 microcontroller, the schematic for which may be found at
;;; olimex.com/Products/PIC/Development/PIC-IO/resources/PIC-IO_revision_C.pdf
;;;
;;;
;;;      U
;;;      +-----+
;;;      | OUT2_  | 1 (RA2)  (RA1) 18 | _OUT3
;;;      | OUT1_  | 2 (RA3)  (RA0) 17 | _OUT4
;;;      | IN1_   | 3 (RA4)  (RA7) 16 | _OSC1
;;;      | /MCLR_ | 4 (RA5)  (RA6) 15 | _OSC2
;;;      | GND_   | 5              14 | _VDD
;;;      | IN2_   | 6 (RB0)  (RB7) 13 | _PGD (ICSP pin 4)
;;;      | TXH = RXD_ | 7 (RB1)  (RB6) 12 | _PGC (ICSP pin 5)
;;;      | RXH = TXD_ | 8 (RB2)  (RB5) 11 | _HBEAT LED (on timer 0)
;;;      | IN3_   | 9 (RB3)  (RB4) 10 | _IN4 (ICSP pin 6)
;;;      +-----+
;;;
PORT1 equ PORTA<<3
OPT01 equ RA4
PORT2 equ PORTB<<3
OPT02 equ RB0
PORT3 equ PORTB<<3
OPT03 equ RB3
PORT4 equ PORTB<<3
OPT04 equ RB4
HBEAT equ RB5

#ifdef LATA
RPORT equ LATA<<3
RHIGH equ LATA>>8
#else
RPORT equ PORTA<<3
RHIGH equ PORTA>>8
#endif
RELAY1 equ RA3
RELAY2 equ RA2
RELAY3 equ RA1
RELAY4 equ RA0

;;; this board uses an 18-pin PIC with an external crystal to watch four opto-
;;; isolators and drive four relays; running this example zOS application each
;;; input/output pair (numbered 1 to 4, coinciding with its job) runs in its own
;;; copy of the relay() re-entrant function and its re-entrant ISR counterpart
;;; optoisr() to reflect respectively the commanded output state from its odd-
;;; numbered global to the relay and input state from the optoisolator into the
;;; even-numbered global:
RLY1OUT equ 0x72
OPT1IN  equ 0x73

```

```

RLY2OUT equ    0x74
OPT2IN  equ    0x75
RLY3OUT equ    0x76
OPT3IN  equ    0x77
RLY4OUT equ    0x78
OPT4IN  equ    0x79
ALL_IOC equ    0x7a            ; logical OR of all IOC flags to watch rise/fall
TMP_IOC equ    0x7b            ; scratch var (globals for init loop then job 5)

;;; the fifth available job is intended to be the monitor application with which
;;; the board can be controlled directly, replaced with a custom application via
;;; the zOS_EXE system call, or for killing relay tasks that are not used and
;;; thus freeing space

;;; uncomment to reduce zOS footprint by 100 words (at cost of zOS_FRK/EXE/FND):
; zOS_MIN      equ    1

        include zos.inc
        include zosmacro.inc

;;; uncomment to pre-load stack positions with indices (for debugging xOS_ROL):
;
; zOS_DBG

;; software interrupt lines used: SI3 to print chars to console, SI4 for RA4 IOC
OUTCHAR equ    zOS_SI3
NON_IOC  equ    zOS_SI4

        pagesel main
        goto    main

input2w macro
    movf    OPT1IN,w            ;inline uint8_t input2w() { // AND of all inputs
    andwf   OPT2IN,w            ; // since an all-zero register means task unrun
    andwf   OPT3IN,w            ; return OPT1IN & OPT2IN & OPT3IN & OPT4IN;
    andwf   OPT4IN,w            ;}
    endm

w2port macro
    andlw   0xf8                ;inline uint8_t* w2port(uint8_t w) {
    xorlw   PORTA<<3            ; return ((w & 0xf8) == ((PORTA<<3) & 0xf8)) ?
    movlw   low PORTA           ; PORTA :
    btfss   STATUS,Z            ; PORTB;
    movlw   low PORTB           ;}
    endm

w2bit macro    file
    andlw   0x07                ;inline uint8_t w2bit(uint8_t* file,
    bsf     STATUS,C            ; uint8_t w) {
    clrf    file                ;
    brw     ;                    ;
    rrf     file,f              ;
    rrf     file,f              ;
    rrf     file,f              ;
    rrf     file,f              ;
    rrf     file,f              ;
    rrf     file,f              ;
    rrf     file,f              ; *file = 1 << (w &= 0x07);
    rrf     file,f              ; return w;
    rrf     file,f              ;}
    endm

myoptol
    addlw   0-1                ;uint8_t myoptol(uint8_t w) { switch (w) {
myopto
    andlw   0x03                ; case 1: return (PORTA<<3) | RA4;
    brw     ; case 2: return (PORTB<<3) | RB0;
    retlw   PORT1|OPTO1          ; case 3: return (PORTB<<3) | RB3;
    retlw   PORT2|OPTO2          ; case 4: return (PORTB<<3) | RB4;

```

```

retlw PORT3|OPT03 ; } // undefined for w < 1 or w > 4
retlw PORT4|OPT04 ; }

myrelay1
addlw 0-1 ;uint8_t myrelay1(uint8_t w) { switch (w) {
myrelay
andlw 0x03 ; case 1: return (PORTA<<3) | RA3;
brw ; case 2: return (PORTA<<3) | RA2;
retlw RPORT|RELAY1 ; case 3: return (PORTA<<3) | RA1;
retlw RPORT|RELAY2 ; case 4: return (PORTA<<3) | RA0;
retlw RPORT|RELAY3 ; } // undefined for w < 1 or w > 4
retlw RPORT|RELAY4 ; }

mychan1
addlw 0-1 ;uint8_t mychan1() { switch (w) {
mychan
andlw 0x03 ; case 1: return 1<<3;
brw ; case 2: return 1<<2;
retlw 0x08 ; case 3: return 1<<1;
retlw 0x04 ; case 4: return 1<<0;
retlw 0x02 ; } // undefined for w < 1 or w > 4
retlw 0x01 ; }

RELAYID equ 0x20 ; PORTA/PORTB/LATA/LATB
OPTOID equ 0x21 ; PORTA/PORTB << 3, 0-7
RELAYP equ 0x22 ; == low LATA/LATB/PORTA/PORTB
OPTOP equ 0x23 ; == low PORTA/PORTB
RELAYB equ 0x24 ; 128/64/32/16/8/4/2/1 to mask with the input
OPTOB equ 0x25 ; 128/64/32/16/8/4/2/1 to mask with the output
OPTOCUR equ 0x26
OPTOLST equ 0x27
MYMASK equ 0x28
SAID_HI equ 0x29

optoisr
zOS_MY2 FSR0
movlw 1[FSR0] ;__isr void optoisr(uint8_t zos_job) {
btfss STATUS,Z ; uint8_t* fsr0 = 0x70 | (bsr<<1); // output var
bra optordy ; if (1[fsr0]) { // initialization has completed
zos_ret

optordy
movf zOS_JOB,w ;
movwf BSR ; bsr = zos_job; // make sure we see our own var
movf OPTOP,w ; uint8_t fsr1 = (relayp == PORTA & 0xff) ?
movwf FSR1L ; &PORTA : &PORTB;
movlw RHIGH ; // 0xff & (this input & mask)
movwf FSR1H ;
movf zOS_AR0,w ; w = zOS_AR0; // in case of a software interrupt
movf zOS_MSK,f ; if (zos_MSK == 0) {
btfss STATUS,Z ; if (INTCON & 1<<IOCIF == 0)
bra optoswi ; zOS_RET(); // not an IOC, maybe timer0 ovf.
btfsc INTCON,IOCIF ;
bra optohwi ; bsr = &IOCBF >> 7;
zos_ret

optohwi
movf OPTOB,w ; w = OPTOB; // our job's single bit of interest
banksel IOCBF
andwf IOCBF,w ; w &= IOCBF; // mask for the port bits
btfss STATUS,Z ; if (w) { // our opto is (at least 1) trigger
bra optoioc ; zOS_MSK = w; // use as scratch var for zero
zos_ret

optoioc
movwf zOS_MSK ; IOCBF ^= w; // clear the IOC flag
xorwf IOCBF,f ; } else

optoswi
andwf INDF1,w ; zOS_RET(); // probably belongs to other job
btfsc STATUS,Z ; }
bra opto_lo ; 1[FSR0] = (w & *fsr1) ? 0xff : ~zos_MSK;

opto_hi
movlw 0xff ; if (all_ioc) { // console out has been initied
movwi 1[FSR0] ; zOS_ARG(0,(w & *fsr1) ? 'H' : 'L');
movlw 'H' ; zOS_TAI(OUTCHAR);
bra optoclr ; // zOS_RFI() implicitly done after zOS_TAI()

opto_lo
comf zOS_MSK,w ; }
movwi 1[FSR0] ;
movlw 'L' ; }

optoclr
movf ALL_IOC,f ; zOS_RET();
btfsc STATUS,Z ; }
bra optodon
zos_arg 0
zos_tai OUTCHAR

optodon
zos_ret

relay
zos_nam "opto+relay pair"

relay
decf zOS_ME ;void relay(void) { // 1<= bsr (job#) <= 4
pagesel myrelay
call myrelay ;
movwf RELAYID ;

w2port
movwf RELAYP ; static uint8_t relayid = myrelay1(bsr);
movf RELAYID,w ; static uint8_t relayp = w2port(relayid);
w2bit RELAYB

decf zOS_ME ; static uint8_t relayb = w2bit(relayid);
pagesel myopto
call myopto ;
movwf OPTOID ; static uint8_t optoid = myoptol(bsr);

w2port
movwf OPTOP ; static uint8_t optop = w2port(optoid);
movf OPTOID,w ; static uint8_t optob = w2bit(optoid);
w2bit OPTOB
movlw 0xff ; // force an initial mismatch
movwf OPTOLST ; static uint8_t optolst = 0xff; // used for RA4

pagesel mychan
decf zOS_ME ;
call mychan ; static uint8_t mymask = mychan1(bsr);
movwf MYMASK ;
zos_swi zOS_YLD ; zOS_SWI(zOS_YLD); // encourage others to init
clrfs SAID_HI ; said_hi = 0;

relayin
zos_my2 FSR0
movf RELAYP,w ; relayin: uint8_t* fsr0 = 0x70 | (bsr << 1);
movwf FSR1L ; uint8_t* fsr1;
movlw RHIGH ;
movwf FSR1H ; fsr1 = (relayp==LATA&0xff) ? &LATA : &LATB;

movlw 0xff ;
movwi 1[FSR0] ; 1[fsr0] = 0xff; // bits nonzero indicates init

relaylp
movf SAID_HI,w ; do {
brw ; if (!said_hi && // haven't announced self yet

relayhi
movf ALL_IOC,f ; all_ioc) { // and job 5 running zOS_CON()
btfsc STATUS,Z ; said_hi = !said_hi;
bra relayrd ; zOS_ADR(fsr0 = &greet);
movlw relayrd-relayhi ; zOS_STR(OUTCHAR);
movwf SAID_HI ; zOS_ARG(0,0);
clrws ; zOS_ARG(1,bsr);
zos_arg 0
movf zOS_ME ; zOS_SWI(OUTCHAR); // "01", "02", "03" or "04"

```

```

        zOS_ARG 1
        zOS_SWI OUTCHAR
;        bra      relayin          ; goto relayin; // to restore FSRs after print
relayrd  movf      MYMASK,w        ;
        andwf     INDF0,w        ; }
        btfscc    STATUS,Z        ;
        bra       relay0         ;
        movf      RELAYB,w        ; if (*fsr0 & mymask)

#ifdef CAUTIOUS
        iorwf     INDF1,w        ; *fsr1 |= relayb; // commanded to 1 by global
        bra       relayop        ;
relay0   comf      RELAYB,w        ; else
        andwf     INDF1,w        ; *fsr1 &= ~relayb; // commanded to 0 by global
relayop  movwf     INDF1          ; // avoid R-M-W issues
#else
        iorwf     INDF1,f        ; *fsr1 |= relayb; // commanded to 1 by global
        bra       relayop        ;
relay0   comf      RELAYB,w        ; else
        andwf     INDF1,f        ; *fsr1 &= ~relayb; // commanded to 0 by global
relayop  #endif
        movf      OPTOP,w        ;
        xorlw     low PORTA      ;
        btfscc    STATUS,Z        ;
        bra       relayld        ; if (OPTOP == PORTA) { // watch in tight loop
;        clrw      PORTA          ; clrwdt(); // shouldn't need to do this
        movlw     low PORTA      ;
        movwf     FSR1L          ;
        movlw     high PORTA     ;
        movwf     FSR1H          ; fsr1 = PORTA;
        movf      INDF1,w        ; if ((OPTOCUR = *fsr1 & OPTOB) != OPTOLST) {
        andwf     OPTOB,w        ; OPTOLST = OPTOCUR;
        movwf     OPTOCUR        ; zOS_SWI(NON_IOC /* or equivalently 0xff */);
        xorwf     OPTOLST,w      ; }
        btfscc    STATUS,Z        ; fsr1 = (relayp==LATA&0xff) ? &LATA : &LATB;
        bra       relayin        ; } else
        xorwf     OPTOLST,f      ; zOS_SWI(zOS_YLD); // PORTA task never yields
        zOS_SWI   NON_IOC        ;
        bra       relayin        ; } while (1);

relayld  zOS_SWI   zOS_YLD
        bra       relaylp        ;}

main     clr      ALL_IOC        ;void main(void) {
        clrf      ALL_IOC        ; volatile uint_8t all_ioc = 0; //job 5 clobbers

create   pagesel   myopto        ; for (w = 0; w < 4; zOS_LAU(&w)) { //1 job/relay
        call      myopto        ; volatile uint_8t tmp_ioc = myopto(w);
        movwf     TMP_IOC        ;
        zOS_ADR   optoisr,zOS_FLA
        movf      TMP_IOC,w      ; fsr0 = &optoisr;
        andlw     0xf8          ;
        xorlw     PORTA<<3      ; if (tmp_ioc & 0xf8 == (PORTA<<3) & 0xf8)
        btfscc    STATUS,Z        ; zOS_INT(0,NON_IOC); // use a SWI from main()
        bra       use_hwi        ; else { // since Port A has no IOC capability
        zOS_INT   0,NON_IOC      ;
        bra       use_swi        ; all_ioc |= w2bit(tmp_ioc); // Port B use IOC

use_hwi  movf      TMP_IOC,w      ; zOS_INT(1<<IOCIF,0); // though so register it
        w2bit     TMP_IOC        ;
        movf      TMP_IOC,w      ;
        iorwf     ALL_IOC,f      ; }

```

```

        zOS_INT 1<<IOCIF,0
use_swi  zOS_ADR   relay,zOS_UNP
        zOS_LAU   WREG
        zOS_ACT   FSR0

        btfscc    WREG,2        ; fsr0 = &relay 0x7fff; // relay() unpriv'ed
        bra       create        ; }

        sublw     zOS_NUM-1      ;
        btfscc    WREG,7        ; if (w == zOS_NUM) // no job remains for zOS_MON
        reset     ; reset();

#ifdef __DEBUG
        banksel   OSCCON
        bsf       OSCCON,IRCF3   ; // change from 0.5MHz default to 16MHz
        movlb     0              ;

CLKRAT   equ      .016000000/.000009600
#else
CLKRAT   equ      .020000000/.000009600
#endif

#ifdef zOS_MIN
        zOS_CON   0,CLKRAT,PIR1,PORTB,RB5,0
#else
        zOS_MAN   0,CLKRAT,PIR1,PORTB,RB5,0
        movlw     OUTCHAR        ; zOS_MON(/*UART*/1,20MHz/9600bps,PIR1,PORTB,5);
        movwi     0[FSR0]        ; zOS_ARG(3, OUTCHAR/*only 1 SWI*/);
#endif

#if 0
        banksel   IOCBP
        movf      ALL_IOC,w      ;
        movwf     IOCBP          ; IOCBP = all_ioc; // IOCIF senses rising optos
        movwf     IOCBN          ; IOCBN = all_ioc; // IOCIF senses falling optos
        bsf       INTCON,IOCIE    ; INTCON |= 1<<IOCIE; // enable edge sensing HWI
        clrf      ALL_IOC        ; ALL_IOC = 0; // will go nonzero once zOS_CON()
#endif

        banksel   TRISA
        bsf       TRISA,RA7       ; TRISA = 0xb0;
        bcf       TRISA,RA6       ; // xtal <-----startup error? race cond'n?
        bsf       TRISA,RA5       ; // MCLR
        bsf       TRISA,OPT01     ; // RA4 is I1
        bcf       TRISA,RELAY1    ; // RA3 is O1
        bcf       TRISA,RELAY2    ; // RA2 is O2
        bcf       TRISA,RELAY3    ; // RA1 is O3
        bcf       TRISA,RELAY4    ; // RA0 is O4
        bsf       TRISB,RB7       ; TRISB = 0xdb;
        bsf       TRISB,RB6       ; // ICSP
        bcf       TRISB,HBEAT     ; // RB5 is HBEAT
        bsf       TRISB,OPT04     ; // RB4 is I4
        bsf       TRISB,OPT03     ; // RB3 is I3
        bcf       TRISB,RB2       ; // RB2 is TXD
        bsf       TRISB,RB1       ; // RB1 is RXD
        bsf       TRISB,OPT02     ; // RB0 is I2

        banksel   ANSELA
        clrf      ANSELA         ; ANSELA = 0x00; // no analog
        clrf      ANSELB         ; ANSELB = 0x00; // no analog

        banksel   OPTION_REG
        bcf       OPTION_REG,T0CS ; OPTION_REG &= ~(1<<TMR0CS); // off Fosc not pin
        bcf       OPTION_REG,PSA  ; OPTION_REG &= ~(1<<PSA); // using max prescaler

        zOS_RUN   INTCON,INTCON   ; zOS_RUN(/*T0IE in*/INTCON, /*T0IF in*/INTCON);
        end        ;}

```

```

;;; zos.inc
;;; a lightweight, small-footprint, preemptively multitasking RTOS for Microchip
;;; Technology's entire enhanced midrange 8-bit PIC microcontroller family:
;;;
;;; jobs (up to 5) are never allowed to manipulate the BSR directly, as that is
;;; the prerogative of zOS (it being used as the current job #) and the bank may
;;; never end up greater than zOS_NUM in user space with interrupts enabled!!!

;;; memory footprint:
;;; ~613 14-bit words for base RTOS i.e. main() starts at 0x0263
;;; ~511 words if zOS_MIN is defined to omit FRK/EXE/FND (thus SWI#4~7=zOS_YLD)
;;;
;;; SRAM footprint:
;;; 86 bank-0 bytes claimed by RTOS, 30 bytes of stack scratch space relocatable
;;;
;;; available bytes      possible jobs with      local bytes/job (+any heap, besides
;;; on PIC device        80 bytes RAM each        2 global bytes) if zOS_NUM set to 5
;;; =====
;;;      128              0                      0 (+2)
;;;      256              1                      0 (+130)
;;;      384              3                      0 (+258)
;;;      512              4                      0 (+386)
;;;      768              5                      80 (+242)
;;;     1,024             5                      80 (+498)
;;;     2,048             5                      80 (+1522)
;;;     4,096             5                      80 (+3570)

;;; you may redefine a constant zOS_NUM with the maximum job number (<6,
;;; as determined by where the general purpose register memory stops, as
;;; the guaranteed 2 bytes global memory isn't sufficient for most jobs)
#ifdef zOS_NUM
#else
zOS_NUM set      5
#endif

;;; you may redefine the location of the scratch space for restoring the stack
;;; after each context switch (by default it is 0x20 in bank zOS_NUM+1, but can
;;; be pulled in on small devices into unused local storage or pushed out if necc
#ifdef zOS_STK
#else
zOS_STK set      (((zOS_NUM+1)<<7)|0x20)
#endif
#ifdef zOS_FRE
#else
zOS_FRE set      (0x2000+((zOS_NUM+1)*0x50)+(0x001e))
#endif

;;; software interrupt infrastructure zOS is based on (even with interrupts off)

;;; 5 user-definable software interrupt lines:
zOS_SB7 equ      7
zOS_SI7 equ      (1<<zOS_SB7)
zOS_SB6 equ      6
zOS_SI6 equ      (1<<zOS_SB6)
zOS_SB5 equ      5
zOS_SI5 equ      (1<<zOS_SB5)
zOS_SB4 equ      4
zOS_SI4 equ      (1<<zOS_SB4)
zOS_SB3 equ      3
zOS_SI3 equ      (1<<zOS_SB3)

;;; 7 system software interrupts for job management:
zOS_FND equ      0x07      ; find a running job <=AR2 by its handle AR1:AR0
zOS_EXE equ      0x06      ; replace this job with a new job (unpriv'ed)
zOS_FRK equ      0x05      ; copy a running job into a new job
zOS_YLD equ      0x04      ; (in)voluntarily cede processor before next irq
zOS_RST equ      0x03      ; restart job at its start address (vs. END+NEW)
zOS_END equ      0x02      ; job killed, slot# available for NEW
zOS_SLP equ      0x01      ; indicate job waiting on its ISR, so don't run

```

```

zOS_NEW equ      0x00      ; create a job (FSR0==addr,AR1:0==isr,AR3:2==IM)

;;; global memory space for 2 scratch registers plus message-passing mailboxes
zOS_JOB equ      0x70      ; next job to run (0 if unknown)
zOS_MSK equ      0x71      ; masked-off software interrupt for ISR to handle
zOS_J1L equ      0x72      ; (repurposeable as scratch after zOS_RFS call)
zOS_J1H equ      0x73
zOS_J2L equ      0x74
zOS_J2H equ      0x75
zOS_J3L equ      0x76
zOS_J3H equ      0x77
zOS_J4L equ      0x78
zOS_J4H equ      0x79
zOS_J5L equ      0x7a
zOS_J5H equ      0x7b
; must disable interrupts e.g. with zOS_ARG(0) before writing SWI args:
zOS_AR0 equ      0x7c
zOS_AR1 equ      0x7d
zOS_AR2 equ      0x7e
zOS_AR3 equ      0x7f

;;; job/shadow register offsets from zOS_J0M, zOS_J1M,...
zOS_HDL equ      0x00      ; handle, the start address of the job
zOS_HDH equ      0x01      ;
zOS_PRB equ      7        ; MSB of HDH indicates privilege(manage others)
zOS_RAM equ      0        ;
zOS_FLA equ      1        ;
zOS_UNP equ      0        ;
zOS_PCL equ      0x02      ; address to resume execution
zOS_PCH equ      0x03      ; "impossible" PCH 0x00==not runnable
zOS_WAI equ      7        ; MSB of PCH indicates sleeping (wait for int)
zOS_SST equ      0x04      ; shadow STATUS
zOS_SWR equ      0x05      ; shadow WREG
zOS_SSP equ      0x06      ; STKPTR to be restored (BSR implied by base)
zOS_SPH equ      0x07      ; PCLATH to be restored
zOS_SF0 equ      0x08      ; shadow FSR0
zOS_SF1 equ      0x0a      ; shadow FSR1
zOS_ISR equ      0x0c      ; interrupt service routine address for the job
zOS_ISH equ      0x0d      ; interrupt service routine address for the job
zOS_HIM equ      0x0e      ; mask for hardware interrupts to process (0=no)
zOS_SIM equ      0x0f      ; mask for software interrupts (low 3 always==1)

zOS_TOS equ      0x0e      ; STKPTR for full stack (0x0f reserved for ISRs)
zOS_BOS equ      0x0b      ; STKPTR for empty stack (first push is to 0x0c)

;;; bank 0 memory space for managing jobs, 1@0x20, 2@0x30, ... , 5@0x60
zOS_J1M equ      0x20
zOS_J2M equ      0x30
zOS_J3M equ      0x40
zOS_J4M equ      0x50
zOS_J5M equ      0x60

zOS_MEM macro     fsrnum,job,offset
local    fsrn
if (fsrnum & 3)
fsrn set 1
else
fsrn set 0
endif
swapf    job,w      ;inline void zOS_MEM(int8_t* *fsrnum,
addlw    0x10        ;      const int8_t* job,
andlw    0x70        ;      const
if (offset)
addlw    offset      ;      int8_t offset) {
endif
movwf    FSR#v(fsrn)L ; *fsrnum = (((job + 1) & 0x07) << 4) + offset;
clrf     FSR#v(fsrn)H ; } // zOS_MEM()
endm

```

```

;;; macro to wind the circular stack around from the running job# to the new job
;;; (before restoring the new job's STKPTR and copying its return address there)
;;; typically: zOS_ROL BSR_SHAD,JOB_NUM(BSR?),zOS_TMP,FSR0,zOS_STK
;;; note: caller is responsible for making sure the STKPTR/_SHAD bank is active
zOS_ROL macro    old,new,temp,fsrnum,base
    local fsrn,loop1,loop2,done
    if (fsrnum & 3)
fsrn set 1
    else
fsrn set 0
    endif
    movlw    low base        ;inline void zOS_ROL(const int8_t* old,
    movwf    FSR#v(fsrn)L    ;          const int8_t* new,
    movlw    high base       ;          int8_t* temp,
    movwf    FSR#v(fsrn)H    ;          int16_t* *fsrnum,
    movf     new,w           ;          int8_t* base) {
    subwf    old,w           ; //responsibility of caller to banksel STKPTR
    btfsc    STATUS,Z        ; if (*new == *old) // nothing to do
    bra      done            ; return;
    decf     WREG,w          ; w = new - old - 1;
    btfsc    WREG,7          ; // set STKPTR to the current location of the
    addlw    5               ; // stack cell that needs to be rotated into
    movwf    STKPTR          ; // STK_TOP, then record this value in temp for
    lslf     STKPTR,f        ; // comparison to know when to exit the loop
    addwf    STKPTR,w        ; // that copies the entire stack (except 0x0f)
    addlw    2               ; // into 30-byte scratch in the unrolled order
    movwf    STKPTR          ;
    movwf    temp           ; for (STKPTR = *temp = 2+3*((w<0) ? (w+5) : w);

loop1
    movf     TOSL,w          ; STKPTR != *temp + 1;
    movwi    FSR#v(fsrn)++   ; STKPTR = (STKPTR>0) ? (STKPTR-1):zOS_TOS;
    movf     TOSH,w          ;
    movwi    FSR#v(fsrn)++   ; (*fsrnum)++ = (TOSH << 8) | TOSL;
    decf     STKPTR,f        ;
    movlw    zOS_TOS         ;
    btfsc    STKPTR,4        ;
    movwf    STKPTR          ;
    movf     temp,w          ;
    xorwf    STKPTR,w        ;
    btfss    STATUS,Z        ; // now rebuild the unrolled stack
    bra      loop1           ;
    clrf     STKPTR          ; for (STKPTR = 0;

loop2
    movi     --FSR#v(fsrn)   ; STKPTR <= zOS_TOS;
    movwf    TOSH            ; STKPTR++) {
    movi     --FSR#v(fsrn)   ; TOSH = *(*fsrnum) >> 8;
    movwf    TOSL            ; TOSL = **--(*fsrnum) & 0x00ff;
    incf     STKPTR,w        ; }
    movwf    STKPTR          ;
    sublw    zOS_TOS         ;
    btfss    WREG,7          ;
    bra      loop2           ; } // zOS_ROL()

done
endm

#ifdef GPASM
zOS_RTL equ    (STATUS_SHAD-FSR1H_SHAD-2)
zOS_RTH equ    (STATUS_SHAD-FSR1H_SHAD-1)
zOS_RTS equ    (STATUS_SHAD-FSR1H_SHAD+2)
#else
zOS_RTL equ    ((STATUS_SHAD-FSR1H_SHAD-2)&0x3f)
zOS_RTH equ    ((STATUS_SHAD-FSR1H_SHAD-1)&0x3f)
zOS_RTS equ    ((STATUS_SHAD-FSR1H_SHAD+2)&0x3f)
#endif

;;; running job#: 1      2      3      4      5
;;; stack pos 15: 3rd(1) 3rd(2) 3rd(3) 3rd(4) 3rd(5)
;;; stack pos 14: 2nd(1) 2nd(2) 2nd(3) 2nd(4) 2nd(5)
;;; stack pos 13: 1st(1) 1st(2) 1st(3) 1st(4) 1st(5)

```

```

;;; stack pos 12: 0th(1) 0th(2) 0th(3) 0th(4) 0th(5)
;;; stack pos 11: 2nd(5) 2nd(1) 2nd(2) 2nd(3) 2nd(4)
;;; stack pos 10: 1st(5) 1st(1) 1st(2) 1st(3) 1st(4)
;;; stack pos 9: 0th(5) 0th(1) 0th(2) 0th(3) 0th(4)
;;; stack pos 8: 2nd(4) 2nd(5) 2nd(1) 2nd(2) 2nd(3)
;;; stack pos 7: 1st(4) 1st(5) 1st(1) 1st(2) 1st(3)
;;; stack pos 6: 0th(4) 0th(5) 0th(1) 0th(2) 0th(3)
;;; stack pos 5: 2nd(3) 2nd(4) 2nd(5) 2nd(1) 2nd(2)
;;; stack pos 4: 1st(3) 1st(4) 1st(5) 1st(1) 1st(2)
;;; stack pos 3: 0th(3) 0th(4) 0th(5) 0th(1) 0th(2)
;;; stack pos 2: 2nd(2) 2nd(3) 2nd(4) 2nd(5) 2nd(1)
;;; stack pos 1: 1st(2) 1st(3) 1st(4) 1st(5) 1st(1)
;;; stack pos 0: 0th(2) 0th(3) 0th(4) 0th(5) 0th(1)

;;; continue with next iteration of HWI-searching loop (mustn't clobber FSR0!)
;;; when searching for the correct hardware interrupt handler, without stack hit
zOS_RET macro
    pagesel zos_nhw
    goto    zos_nhw          ;#define zOS_RET() goto zos_nhw
endm

;;; at the end of any interrupt handler goes back to scheduler without stack hit
zOS_RFI macro
    pagesel zos_noc
    goto    zos_noc          ;inline void zOS_RFI(void) { goto zos_noc; }
endm

zOS_RFS macro    retreg
    pagesel zos_sch          ;inline void zOS_RFS(int8_t* retreg) { //from SWI
    if (retreg-WREG)
        movf    retreg,w      ; w = *retreg; goto zos_sch; //clobbers WREG_SHAD
    endif
    goto    zos_sch          ; } // zOS_RFS()
endm

;;; find something runnable (i.e. PCH != 0, but sleep MSB is OK), at job+/-1
;;; according to incr then branch to unf if job-1 == 0 or job+1 > zOS_NUM,
;;; with fsrnum pointing to job's bank 0 structure and then incremented +/-16
zOS_LIV macro    fsrnum,job,incr,unf
    local fsrn,loop
    if (fsrnum & 3)
fsrn set 1
    else
fsrn set 0
    endif
loop
    if (incr)
        movlw    0x10          ;inline int8_t zOS_LIV(int8_t* *fsrnum,
    else
        movlw    0-0x10        ; uint8_t *job, int8_t incr, void *(*unf)()) {
    endif
    addwf    FSR#v(fsrn)L,f    ; do {
    if (incr)
        incf     job,f          ; *fsrnum += incr ? 0x10 : -0x10; // next struct
        movlw    0xff-zOS_NUM   ; job += incr ? 1 : -1; // next job#
        addwf    job,w          ; if ((job == 0) || (job >= zOS_NUM+1)) { //past
        btfss    WREG,7          ;
    else
        decf     job,f          ; goto unf; // Z was set
        btfsc    STATUS,Z        ; } else if (zOS_PCH[fsrnum]) // found runnable
    endif
    bra      unf              ; return w = zOS_PCH[fsrnum]; // Z was cleared
    movi     zOS_PCH[FSR#v(fsrn)]
    btfsc    STATUS,Z          ; } while (1); // job is runnable (or unf was 0)
    bra      loop              ; } // zOS_LIV()
endm

#ifdef FSR0
#else

```

```

FSR0     equ     FSR0L
#endif
#ifdef FSR1
#else
FSR1     equ     FSR1L
#endif

;; a job switch is attempted with every incoming interrupt
;; user jobs are responsible for processing their own interrupts
;; with an interrupt handler registered at the time of creation

org      0x0000
pagesel  zos_ini
goto     zos_ini      ;<--zos_ini is run upon reset to bootstrap zOS

org      0x0002
pagesel  zos_swj
goto     zos_swj      ;<--zos_SWI is call to 0x0002, a jump to zos_swj

;; enter handler which will zOS_RFI() to zos_sch if it's the correct one
;; (and we're not still in the bank-0 initialization before interrupts),
;; after clearing the interrupt flag...else zOS_RET() back up to zos_nhw

org      0x0004
;; find first willing handler for an enabled interrupt matching _xIM bit
#ifdef PIE0
zos_PIE  equ     PIE0
#else
zos_PIE  equ     INTCON
#endif
zos_004
movlw    zOS_NUM+1      ;__isr void zos_004(void) {
movwf    zOS_JOB        ; zOS_JOB = zOS_NUM+1; // search from high to low
zos_MEM  FSR0,zOS_JOB,0 ; fsr0 = 0x10 * (1 + zOS_JOB);

zos_nhw
zos_LIV  FSR0,zOS_JOB,0,zos_004
clrwdt   ; do { // until serviceable by running ISR since
banksel  zos_PIE
movlw    zOS_HIM[FSR0]  ; int8_t w = 0; // no runnable job schedulable
andwf    zos_PIE,w      ; clrwdt();
btfss    STATUS,Z       ; while (zos_LIV(&fsr0, &zOS_JOB, 0)) {
bra       zos_cmp       ; //match enabled interrupts against HIM fields

#ifdef PIE1
movlw    zOS_HIM[FSR0]  ; if ((w = zOS_HIM[fsr0] & zOS_PIE))
banksel  PIE1
andwf    PIE1,w         ; break;
btfss    STATUS,Z       ; if ((w = zOS_HIM[fsr0] & zOS_PIE1))
bra       zos_cmp       ; break;

#endif
#ifdef PIE2
movlw    zOS_HIM[FSR0]  ;
andwf    PIE2,w         ;
btfss    STATUS,Z       ; if ((w = zOS_HIM[fsr0] & zOS_PIE2))
bra       zos_cmp       ; break;

#endif
#ifdef PIE3
movlw    zOS_HIM[FSR0]  ;
andwf    PIE3,w         ;
btfss    STATUS,Z       ; if ((w = zOS_HIM[fsr0] & zOS_PIE3))
bra       zos_cmp       ; break;

#endif
#ifdef PIE4
movlw    zOS_HIM[FSR0]  ;
andwf    PIE4,w         ;
btfss    STATUS,Z       ; if ((w = zOS_HIM[fsr0] & zOS_PIE4))
bra       zos_cmp       ; break;

#endif
#ifdef PIE5
movlw    zOS_HIM[FSR0]  ;

```

```

andwf    PIE5,w         ;
btfss    STATUS,Z       ; if ((w = zOS_HIM[fsr0] & zOS_PIE5))
bra       zos_cmp       ; break;

#endif
#ifdef PIE6
movlw    zOS_HIM[FSR0]  ;
andwf    PIE6,w         ;
btfss    STATUS,Z       ; if ((w = zOS_HIM[fsr0] & zOS_PIE6))
bra       zos_cmp       ; break;

#endif
#ifdef PIE7
movlw    zOS_HIM[FSR0]  ;
andwf    PIE7,w         ;
btfss    STATUS,Z       ; if ((w = zOS_HIM[fsr0] & zOS_PIE7))
bra       zos_cmp       ; break;

#endif
#ifdef PIE8
movlw    zOS_HIM[FSR0]  ;
andwf    PIE8,w         ;
btfss    STATUS,Z       ; if ((w = zOS_HIM[fsr0] & zOS_PIE8))
bra       zos_cmp       ; break;

#endif
#ifdef PIE9
movlw    zOS_HIM[FSR0]  ;
andwf    PIE9,w         ;
btfss    STATUS,Z       ; if ((w = zOS_HIM[fsr0] & zOS_PIE9))
bra       zos_cmp       ; break; // found a potential handler for any

#endif
bra       zos_nhw       ; } // interrupt flag in this bit position
zos_cmp
clrf     zOS_MSK        ; if (w) {
movlw    zOS_ISH[FSR0]  ; zOS_MSK = 0; //indicates HWI (not SWI) type
movwf    PCLATH         ; *(zos_ISR[fsr0]);
movlw    zOS_ISR[FSR0]  ; }
movwf    PCL             ; } // if handler refuses, loops to the next job

;; scheduler begins here, called either after HWI/SWI done or zOS_RUN():

zos_sch
banksel  WREG_SHAD
movwf    WREG_SHAD      ;zos_sch: // w sent via zOS_RFS()

zos_noc
banksel  WREG_SHAD
movf     BSR_SHAD,w      ; WREG_SHAD = w;zos_noc://lobber from zOS_RFI()
btfsc    STATUS,Z       ; // stay in _SHAD/STKPTR/TOS bank until retfie
bra       zos_don        ; if ((zos_JOB = BSR_SHAD)!= 0)//2x max or '004
movwf    zOS_JOB        ; for (zos_MSK = 2; zOS_MSK; zOS_MSK--) {
movlw    3              ;
movwf    zOS_MSK        ; //zos_MSK=2 first time through,1 after wrap
bra       zos_lst       ; zOS_MEM(fsr0,zOS_JOB,0);

zos_itr
zos_LIV  FSR0,zOS_JOB,1,zos_wra
clrwdt   ; //zos_LIV leaves PCH in WREG, test runnable?
btfsc    WREG,zOS_WAI    ; while(zos_LIV(fsr0,zOS_JOB,1)&(1<zOS_WAI))
bra       zos_itr       ; clrwdt();

;; if this point is reached, a runnable job was found with job# zOS_JOB
;; (but we skip a whole bunch of trivial copies if zOS_JOB==BSR_SHAD)
movf     BSR_SHAD,w      ;
xorwf    zOS_JOB,w       ;
btfsc    STATUS,Z       ;
bra       zos_don        ; if (zos_JOB != BSR_SHAD) {

;; copy the interrupted job's (BSR_SHAD) criticals into its bank 0 slot;
zos_MEM  FSR0,BSR_SHAD,zOS_PCL
movf     TOSL,w          ; fsr0 = 0x10 * (1+BSR_SHAD) + zOS_PCL;
movwi    FSR0++          ; *fsr0++ = TOSL; // return address from IRQ
movf     TOSH,w          ;
movwi    FSR0++          ; *fsr0++ = TOSH;

```

```

movf    STATUS_SHAD,w    ;
movwi    FSR0++,         ;    *fsr0++ = STATUS_SHAD;
movf    WREG_SHAD,w      ;
movwi    FSR0++,         ;    *fsr0++ = WREG_SHAD;
movf    STKPTR,w         ;
movwi    FSR0++,         ;    *fsr0++ = STKPTR; // not BSR_SHAD
movf    PCLATH_SHAD,w    ;
movwi    FSR0++,         ;    *fsr0++ = PCLATH_SHAD;
movf    FSR0L_SHAD,w     ;
movwi    FSR0++,         ;    *fsr0++ = FSR0L_SHAD;
movf    FSR0H_SHAD,w     ;
movwi    FSR0++,         ;    *fsr0++ = FSR0H_SHAD;
movf    FSR1L_SHAD,w     ;
movwi    FSR0++,         ;    *fsr0++ = FSR1L_SHAD;
movf    FSR1H_SHAD,w     ;
movwi    FSR0++,         ;    *fsr0++ = FSR1H_SHAD;

;; by pure chance this clobbers the "unused" range 0x72~0x7b on 1st run!
movlw    0x7c            ;
xorwf    FSR0L,f         ;
btfss    STATUS,Z        ;
bra      zos_no0          ;    if (fsr0 == 0x007c) {
movlw    0x0a            ;
movwf    FSR0H           ;
movlw    0x72            ;
movwf    FSR0L           ;    fsr0 = 0x0072;
clrw     ;                for (uint8_t i; i < 10; i++)

zos_re0
movwi    FSR0++,         ;    *fsr0 = 0;
decfsz   FSR0H,f         ;
bra      zos_re0         ;    }

zos_no0

;; get stack spun around to where zOS_JOB expects it on return from ISR
zos_ROL   BSR_SHAD,zOS_JOB,zOS_MSK,FSR1,zOS_STK

;; copy zOS_JOB's criticals out of its bank 0 slot
zos_MEM   FSR0,zOS_JOB,zOS_SST
movwi    FSR0++,         ;    fsr0 = 0x10 * (1+zOS_JOB) + zOS_SST;
movwf    STATUS_SHAD    ;    STATUS_SHAD = *fsr0++;
movwi    FSR0++,         ;
movwf    WREG_SHAD      ;    WREG_SHAD = *fsr0++;
movf     zOS_JOB,w       ;    //point to correct 80-byte local SRAM page
movwf    BSR_SHAD       ;    BSR_SHAD = zOS_JOB; // not STKPTR
++FSR0    ;                //^^ notice BSR = zOS_JOB upon retfie! ^^
movwf    PCLATH_SHAD    ;    PCLATH_SHAD = **++fsr0;
movwi    ++FSR0         ;
movwf    FSR0L_SHAD     ;    FSR0L_SHAD = **++fsr0;
movwi    ++FSR0         ;
movwf    FSR0H_SHAD     ;    FSR0H_SHAD = **++fsr0;
movwi    ++FSR0         ;
movwf    FSR1L_SHAD     ;    FSR1L_SHAD = **++fsr0;
movwi    ++FSR0         ;
movwf    FSR1H_SHAD     ;    FSR1H_SHAD = **++fsr0;

;; set new job stack pointer, last step before completing context switch
movwi    zOS_RTS[FSR0]  ;
movwf    STKPTR         ;    STKPTR = zOS_SSP[FSR0-11];
movwi    zOS_RTL[FSR0]  ;    TOSL = zOS_PCH[FSR0-11];
movwf    TOSL           ;    TOSH = zOS_PCH[FSR0-11];
movwi    zOS_RTH[FSR0]  ;    return (void)__isr;
movwf    TOSH           ;    }

zos_don
retfie    ;                //if this point is reached, search wrapped:

zos_wra
clrf     zOS_JOB        ;    fsr0 = 0x10 * (1 + (zOS_JOB = 0));

zos_1st
zos_MEM   FSR0,zOS_JOB,0 ;    } // wrap around only once, else wait for IRQ
decfsz   zOS_MSK,f      ;    } while (1); // (since no job is schedulable)

bra      zos_itr         ;} // zOS_004()
bra      zos_004         ;int8_t zos_swj(int8_t w){ // call vector at 002

;; software interrupt processing reached by jumping to 0x0002 with W set
;; which then calls to zos_swj, or by jumping to zos_skp after already
;; processing a previous interrupt (since there is only 1 level of SHAD)
;; to skip the copy into the shadow registers

zos_skp
movwf    zOS_MSK         ;
bra      zos_sk2         ;

zos_swj
;; save the shadow registers (for the ones that have them) to use retfie
bcf     INTCON,GIE       ;    INTCON &= ~(1<<GIE); // interrupt would be bad
movwf    zOS_MSK         ;    zOS_MSK = WREG; // the software interrupt type
movf     STATUS,w        ;
movwf    zOS_JOB         ;    // only convenient temporary global for STATUS
movf     BSR,w           ;
banksel  BSR_SHAD        ;    // BSR = the job# that made the interrupt call
movwf    BSR_SHAD        ;    BSR_SHAD = BSR;
movf     zOS_JOB,w       ;
movwf    STATUS_SHAD     ;    STATUS_SHAD = zos_job = STATUS;
movf     PCLATH,w        ;
movwf    PCLATH_SHAD     ;    PCLATH_SHAD = PCLATH;
movf     FSR0L,w         ;
movwf    FSR0L_SHAD      ;    FSR0L_SHAD = FSR0L;
movf     FSR0H,w         ;
movwf    FSR0H_SHAD      ;    FSR0H_SHAD = FSR0H;
movf     FSR1L,w         ;
movwf    FSR1L_SHAD      ;    FSR1L_SHAD = FSR1L;
movf     FSR1H,w         ;
movwf    FSR1H_SHAD      ;    FSR1H_SHAD = FSR1H;

zos_sk2
;; see if the interrupt type is a system one (<8)
pagesel  zos_swh
movlw    zOS_SI7|zOS_SI6|zOS_SI5|zOS_SI4|zOS_SI3
andwf    zOS_MSK,w       ;    if (0 == /* call-type number: */ WREG_SHAD &
btfss    STATUS,Z        ;    (zOS_SI7|zOS_SI6|zOS_SI5|zOS_SI4|zOS_SI3)) {
goto     zos_swh         ;    // handle a system zOS_SWI call:

;; zOS_NEW requires us to search for a BSR value first among empty slots
movf     BSR_SHAD,w      ;
movwf    BSR             ;    // BSR unchanged from what it had been at call
movf     zOS_MSK,f       ;
btfss    STATUS,Z        ;    if (zOS_MSK == zOS_NEW /*==0*/) {
bra      zos_swp         ;    zos_cre:

zos_cre
clrf     zOS_JOB         ;    zos_job = 0;
zos_MEM   FSR1,zOS_JOB,0

zos_emp
movlw    0x10            ;    for (fsr1 = 0x10*(1+zos_job);
addwf    FSR1L,f         ;
incf     zOS_JOB,f       ;    zos_job++ <= zOS_NUM;
movlw    0xff-zOS_NUM    ;
addwf    zOS_JOB,w       ;
btfsc    STATUS,Z        ;    fsr1 += 0x10) {
bra      zos_err         ;    if (zOS_PCH[FSR1] == 0)
movwi    zOS_PCH[FSR1]   ;    break;
btfss    STATUS,Z        ;    }
bra      zos_emp         ;    if (zos_job <= zOS_NUM) {

zos_dup
movf     FSR0L,w         ;    // save handle now so we can re-use fsr0
movwi    zOS_HDL[FSR1]   ;    // (no harm if we don't validate it as PCH)
movf     FSR0H,w         ;    zOS_HDL[fsr1] = fsr0 & 0x00ff;
movwi    zOS_HDH[FSR1]   ;    zOS_HDH[fsr1] = fsr0 >> 8;
movf     BSR,f           ;    if (bsr == 0)
btfsc    STATUS,Z        ;    goto zos_swk; // job#0 (launcher) has perm
bra      zos_swk         ;    fsr0 = 0x10 * (1+bsr); // struct for caller
zos_MEM   FSR0,BSR,0
movwi    zOS_HDH[FSR0]   ;    if (zOS_HDH[fsr0] & (1<<zOS_PRB))

```

```

        btfsc  WREG,zOS_PRB    ;   goto zos_swk; // job has privileged perms
        bra    zos_swk        ;   }

zos_err
        clrf   zOS_JOB        ;   zos_job = 0;
        zOS_RFS zOS_JOB      ;   zOS_RFS(zOS_JOB); // perms error or no empty

        ;; see if we're not running inside a job context (1 <= job# <= zOS_NUM)
        ;; in which case need to grab the targeted job from AR0 (if not zOS_NEW)
        ;; or find a targetable slot (if zOS_NEW)
        ;; unprivileged jobs can only do most things to themselves

zos_sw0
        movf   BSR,w          ;   } else {
        movwf  zOS_JOB        ;   zos_job = bsr;
        btfsc  STATUS,Z       ;   if (bsr != 0) {
        bra    zos_elv        ;   fsrl = 0x10 * (1+bsr); // struct for job
        zOS_MEM FSR1,BSR,0
        moviw  zOS_HDH[FSR1]   ;   if (zos_HDH[fsrl] & (1<<zOS_PRB) == 0)
        btfss  WREG,zOS_PRB   ;   goto zos_swk; // disallowed job in zOS_AR0
        bra    zos_swk        ;   }

        ;; desired job# (instead of this one) into BSR from AR0 (if not zOS_NEW)

zos_elv
        movf   zOS_AR0,w      ;   // access granted, bring the patient to me
        movwf  BSR            ;   bsr = zOS_AR0;
        zOS_MEM FSR1,BSR,0

zos_swk
        movf   zOS_MSK,w      ;   }
        brw    zos_sw0        ;   switch (zos_MSK) { // guaranteed < 8
        bra    zos_sw1
        bra    zos_sw2
        bra    zos_sw3
        bra    zos_sw4
        bra    zos_sw5
        bra    zos_sw6
        bra    zos_sw7        ;   case zOS_NEW:

zos_sw0
        movf   zOS_AR0,w      ;
        movwi  zOS_ISR[FSR1]   ;   zOS_ISR[fsrl] = zOS_AR0;
        movf   zOS_AR1,w      ;
        movwi  zOS_ISH[FSR1]   ;   zOS_ISH[fsrl] = zOS_AR1;
        movf   zOS_AR2,w      ;
        movwi  zOS_HIM[FSR1]   ;   zOS_HIM[fsrl] = zOS_AR2;
        movf   zOS_AR3,w      ;
        movwi  zOS_SIM[FSR1]   ;   zOS_SIM[fsrl] = zOS_AR3;
        bra    zos_sw3        ;   goto zos_sw3;

zos_sw1
        moviw  zOS_PCH[FSR1]   ;   case zOS_SLP:
        iorlw  0x80            ;   zOS_PCH[fsrl] |= 0x80;
        movwi  zOS_PCH[FSR1]   ;   zOS_RFS(zOS_JOB);
        zOS_RFS zOS_JOB

zos_sw2
        movf   BSR,w          ;   case zOS_END:
        banksel PCLATH_SHAD   ;
        xorwf  BSR_SHAD,w     ;
        btfsc  STATUS,Z       ;   if (bsr == BSR_SHAD) // if killing self wipe
        clrf   TOSH           ;   TOSH = 0; // stack so PC can't get restored
        xorwf  BSR_SHAD,w     ;
        movwf  BSR            ;
        clrw   zos_sw3        ;   zOS_PCH[fsrl] = 0; // so scheduler won't see
        movwi  zOS_PCH[FSR1]   ;   zOS_RFS(zOS_JOB); // killing is so quick
        zOS_RFS zOS_JOB

zos_sw3
        moviw  zOS_HDL[FSR1]   ;   case zOS_RST: zos_sw3:
        movwi  zOS_PCL[FSR1]   ;   // retain HDL MSB (which indicate privilege)
        moviw  zOS_HDH[FSR1]   ;   zOS_PCL[fsrl] = zOS_HDL[fsrl];

#ifdef zOS_AUT
        andlw  0x7f           ;   // clear PC MSB (which indicates sleepiness)
#else

```

```

        iorlw  0x80            ;   // set PC MSB (so must explicitly activate)
#endif
        movwi  zOS_PCH[FSR1]   ;   zOS_PCH[fsrl] = zOS_HDH[fsrl] & 0x7f;
        movlw  zOS_BOS        ;   zOS_SSP[fsrl] = zOS_BOS;
        movwi  zOS_SSP[FSR1]   ;

        lslf   zOS_JOB,w      ;
        iorlw  0x70            ;
        movwf  FSR1L          ;   fsrl = 0x70 | (zos_JOB << 1);
        clrw   zos_sw4        ;   0[fsrl] = 1[fsrl] = 0; // mailbox guar'ed 0
        movwi  0[FSR1]        ;   case zOS_YLD:
        movwi  1[FSR1]        ;   zOS_RFS(zOS_JOB);

zos_sw4
#ifdef zOS_MIN
zos_sw5
zos_sw6
zos_sw7
        zOS_RFS zOS_JOB
#else
        zOS_RFS zOS_JOB
zos_sw5
        ;; copy job BSR's 0x20-0x6f into every non-running bank first
        clrf   FSR1L          ;   case zOS_FRK:
        clrf   FSR1H          ;   fsrl = 1 << 7;
        clrf   zOS_JOB        ;   for (zos_job = 1;

zos_cp1
        movlw  0x80            ;   zos_job++ <= zOS_NUM; fsrl += 0x80) {
        andwf  FSR1L,f         ;   fsrl &= 0xff80;
        addwf  FSR1L,f         ;
        clrw   zos_sw4        ;
        addwfc FSR1H,f         ;   fsrl += 0x80;
        incf   zOS_JOB,f       ;
        movlw  0xff-zOS_NUM    ;
        addwf  zOS_JOB,w       ;
        btfsc  STATUS,Z       ;
        bra    zos_cp2        ;

        zOS_MEM FSR0,zOS_JOB,0
        moviw  zOS_PCH[FSR0]   ;   fsr0 = 0x10 * (1+zOS_JOB);
        btfss  STATUS,Z       ;   if (zos_PCH[fsr0] == 0)
        bra    zos_cp1        ;   continue; // can't touch a running job

        lsr    BSR,w          ;
        movwf  FSR0H           ;
        clrf   FSR0L           ;
        rrf    FSR0L,f         ;
        movlw  0x6f           ;
        iorwf  FSR0L,f         ;   fsr0 = (BSR << 7) | 0x6f;
        iorwf  FSR1L,f         ;   for (fsrl |= 0x6f; fsrl & 0x7f >= 0x20;

zos_cp2
        moviw  FSR0--          ;
        movwi  FSR1--          ;   *fsrl-- = *fsr0--
        movlw  0x60            ;
        andwf  FSR0L,w         ;
        btfss  STATUS,Z       ;
        bra    zos_cp2        ;
        bra    zos_cp1        ;   }

zos_cp3
        ;; now copy job BSR's bank0 struct to the zOS_AR registers and zOS_NEW()
        ;;;FIXME: should copy the rest of state, i.e. memory variables to be a true fork
        ;;;FIXME: disallow fork if any HWI is defined for the process (assume conflicts)
        movf   BSR,w          ;
        movwf  zOS_JOB        ;   zOS_JOB = BSR;
        zOS_MEM FSR1,zOS_JOB,0
        moviw  zOS_PCH[FSR1]   ;   fsrl = zOS_MEM(&fsrl, zOS_JOB, 0);
        btfsc  STATUS,Z       ;
        bra    zos_sw4        ;   if ((w = zOS_PCH[fsrl]) != 0) {

```



```

    moviw  zOS_HDL[FSR1] ;
    movwf  FSR0L         ;
    moviw  zOS_HDH[FSR1] ;
    movwf  FSR0H         ;    fsr0 = (zOS_HDH[fsr1]<<8) | zOS_HDL[fsr1];
    moviw  zOS_ISR[FSR1] ;
    movwf  zOS_AR0       ;    zOS_AR0 = zOS_ISR[fsr1];
    moviw  zOS_ISH[FSR1] ;
    movwf  zOS_AR1       ;    zOS_AR1 = zOS_ISH[fsr1];
    moviw  zOS_HIM[FSR1] ;
    movwf  zOS_AR2       ;    zOS_AR2 = zOS_HIM[fsr1];
    moviw  zOS_SIM[FSR1] ;
    movwf  zOS_AR3       ;    zOS_AR3 = zOS_SIM[fsr1];
    banksel WREG_SHAD
    clrf   WREG_SHAD      ;    WREG_SHAD = zOS_NEW;
    movlb  0              ;    zOS_MSK = 0; //spooof having passed zOS_NEW
    clrf   zOS_MSK        ;    goto zos_cre;//spooof privilege to fork self
    bra    zos_cre        ;    } else zOS_RFS(w);

zos_sw6
    movf   BSR,w          ;    case zOS_EXE:
    movwf  zOS_JOB        ;    zOS_JOB = BSR;
    zOS_MEM FSR1,zOS_JOB,0
    banksel WREG_SHAD     ;    fsr1 = 0x10 * (1+zOS_JOB);
    clrf   WREG_SHAD      ;    WREG_SHAD = zOS_NEW;
    movlb  0              ;    //spooof privilege to overwrite
    bra    zos_dup        ;    goto zos_dup;

zos_sw7
    movf   zOS_AR2,w      ;    case zOS_FND:
    btfss  STATUS,Z       ;
    movlw  zOS_NUM        ;
    addlw  1              ;
    movwf  zOS_JOB        ;
    addlw  0xfe-zOS_NUM    ;    if (zOS_AR2 && ((uint8_t)zOS_AR2<=zOS_NUM))
    btfsc  WREG,7         ;    zOS_JOB = zOS_AR2 + 1;
    movlw  1+zOS_NUM      ;    else
    movwf  zOS_JOB        ;    zOS_JOB = zOS_NUM + 1;
    zOS_MEM FSR1,zOS_JOB,0 ;    fsr1 = 0x10 * (1 + zOS_JOB);

zos_nxt
    zOS_LIV FSR1,zOS_JOB,0,zos_bad
    moviw  zOS_HDL[FSR1] ;    while (zOS_LIV(&fsr1, &zOS_JOB, 0)) {
    xorwf  zOS_AR0,w      ;
    btfss  STATUS,Z       ;
    bra    zos_nxt        ;
    moviw  zOS_HDH[FSR1] ;    void (*a)() = (zOS_AR1<<8)|zOS_AR0;
    xorwf  zOS_AR1,w      ;    void (*b)() = (zOS_HDH[fsr1]<<8)|zOS_HDL[fsr1]
;
    andlw  0x7f           ;
    btfss  STATUS,Z       ;    if (a & 0x7f == b & 0x7f)
    bra    zos_nxt        ;    zOS_RFS(zOS_JOB);
    zOS_RFS zOS_JOB       ;    }

zos_bad
    clrw   ;
    zOS_RFS WREG          ;    zOS_RFS(w = 0);

#endif

; ; else handle the software interrupt with the first registered handler

zos_swh
    banksel BSR_SHAD
    incf   BSR_SHAD,w     ; // a swi number of 0xff is special now, will
    incfsz zOS_MSK,f      ; // cause the calling job to invoke its own
    movlw  1+zOS_NUM      ; // handler without knowledge of its SWI code!
    decf   zOS_MSK,f      ; // (at the cost of 4 extra instruction cycles)
    movwf  zOS_JOB        ; zos_job =1+((zos_msk==0xff)?BSR_SHAD:zOS_NUM);

    zOS_MEM FSR0,zOS_JOB,0 ; while (zOS_LIV(&fsr0, &zOS_JOB, 0)) { //search

zos_sw1
    zOS_LIV FSR0,zOS_JOB,0,zos_swm
    moviw  zOS_SIM[FSR0] ;
    andwf  zOS_MSK,w      ;
    btfsc  STATUS,Z       ;
    bra    zos_sw1        ;

    bra    zos_sw1        ;    if ((zos_msk & zOS_SIM[fsr0]) != 0) { //found
    movwf  zOS_MSK        ;    zos_msk &= zOS_SIM[fsr0];
    moviw  zOS_ISH[FSR0] ;    goto (void*)(zOS_ISR[fsr0]); // will zOS_RFS
    movwf  PCLATH         ; }
    moviw  zOS_ISR[FSR0] ; }
    movwf  PCL            ; zOS_RFS(WREG = 0);

; ; no registered SWI handler: jump into the hardware interrupt scheduler

zos_swm
    zOS_RFS WREG

zos_ini
    ; ; clear out page 0 to reflect no running tasks, set global data to 0's
    movlb  0              ; "invalid" job# used to get perms for zOS_NEW
    movlw  0x7f           ; bsr = 0;
    movwf  FSR0L          ;
    clrf   FSR0H          ; for (fsr0 = 0x007f; fsr >= 0x0020; fsr--)

zos_zer
    clrw   ;
    movwi  FSR0--         ; *fsr = 0; // only zOS_PCH is critical
    movlw  0x60           ;
    andwf  FSR0L,w        ;
    btfss  STATUS,Z       ;
    bra    zos_zer        ;

; ; your program starts here, with a series of launcher instructions for
; ; 1) setting up oscillators, timers, other peripherals, etc.
; ; (with the appropriate and inevitable bank switching)
; ; 2) starting jobs with calls to zOS_NEW or its zOS_LAU wrapper
; ; (being sure to stay in bank 0 or using job macros zOS_CON/zos_MON)
; ; 3) calling zOS_RUN (which will enable interrupts) to start job 1

```

```

;;; zosmacro.inc
;;; potentially useful (but not mandatory) macros for zOS
;;;
;;; total memory footprint (for a PIC16F1847, including the zOS base):
;;; no memory words used upon inclusion (before expansion of a macro)
;;; ~256 14-bit words if only zOS_CON() job is started to buffer console output
;;; _?_ 14-bit words for full-featured monitor zOS_MON()
;;; _?_ 14-bit words for job manager shell zOS_MAN()

#ifdef UCFG
#define zOS_ME BSR,w : xorlw 0x8; // advance zOS use past DPSRAM; FIXME:untested
#else
#define zOS_ME BSR,w ; // "movf/andwf/xorwf zOS_ME" can't clobber BSR
#endif

zOS_GLO macro fsrnum,job
    local fsrn
    if (fsrnum & 3)
fsrn set 1
    else
fsrn set 0
    endif
    if (job)
        lslf job,w ;inline void zOS_GLO(int8_t**fsrnum,int8_t*job){
    else
        lslf zOS_ME ;
    endif
    andlw 0x0e ; int8_t w = 0x70 | ((job ? *job : bsr) << 1);
    iorlw 0x70 ;
    movwf FSR#v(fsrn)L ;// documentation suggests 5 but BSR now 6 bits!
    movlw 0x1f ; *fsrnum = (*fsrnum & 0x1f00) | w;
    andwf FSR#v(fsrn)H,f ;} // zOS_GLO()
endm

zOS_MY2 macro fsrnum ;inline int8_t zOS_MY2(int8_t**fsrnum){
    zOS_GLO fsrnum,0 ; return zOS_GLO(fsrnum, 0);
endm ;} // zOS_MY2()

zOS_LOC macro fsrnum,job,offset
    local fsrn
    if (fsrnum & 3)
fsrn set 1
    else
fsrn set 0
    endif
    if (offset)
        movlw offset<<1 ;inline int8_t zOS_LOC(int8_t* *fsrnum,
        movwf FSR#v(fsrn)L ; int8_t* job, uint8_t offset) {
    else
        clrf FSR#v(fsrn)L ;
    endif
    if (job - FSR#v(fsrn)H)
        lsrif job,w ;
        movwf FSR#v(fsrn)H ; return (*fsrnum = (job<<7) | offset) >> 8;
    else
        lsrif job,f
    endif
    rrf FSR#v(fsrn)L,f ;} // zOS_LOC()
endm

zOS_ADR macro adr,msb
    movlw low adr ;inline void zOS_ADR(void* a) {
    movwf FSR0L ; if (msb) fsr0 = 0x8000 | a;
    movlw high adr ; else fsr0 = 0x7fff & a;
    movwf FSR0H ;} // zOS_ADR()
    if (msb)
        bsf FSR0H,7
    else
        bcf FSR0H,7

```

```

endif
endm

zOS_INT macro lhw,lsw
    if (lhw|lsw)
        movf FSR0L,w ;inline void zOS_INT(const lhw, const lsw) {
        zOS_ARG 0
        movf FSR0H,w ; if (lhw == 0 && lsw == 0) fsr0 = 0;
        zOS_ARG 1
        movlw lhw ; zOS_ARG(0, fsr0 & 0x00ff);
        zOS_ARG 2
        movlw lsw ; zOS_ARG(1, fsr0 >> 8);
        zOS_ARG 3
        else
            clrw ; zOS_ARG(2, lhw);
            movwf FSR0L ; zOS_ARG(3, lsw);
            movwf FSR0H ;} // zOS_INT()
        zOS_ARG 0
        zOS_ARG 1
        zOS_ARG 2
        zOS_ARG 3
    endif
endm

zOS_SWI macro type ;inline void zOS_SWI(const int8_t type) {
    movlw type ;
    movlp 0x00 ; zos_swj(type);
    call 0x02 ;} // zOS_SWI()
endm

zOS_TAI macro type ;inline void zOS_TAI(const int8_t type) {
    movlw type ; w = type; goto zos_skp;
    pagesel zos_skp
    goto zos_skp ;} // zOS_TAI()
endm

zOS_LAU macro stash ;inline void zOS_LAU(int8_t* stash) {
    local retry

retry
    zOS_SWI zOS_NEW
    bcf INTCON,GIE ; do { w = zOS_SWI(zOS_NEW);

#ifdef CAUTIOUS
    movf BSR,f ; INTCON &= ~(1<<GIE); // prevent deadlock
    btfss STATUS,Z ; if (bsr) // arising from an
    bsf INTCON,GIE ; INTCON &= 1<<GIE; // interrupt right now
#endif

    movf WREG,w ;
    btfsc STATUS,Z ;
    bra retry ; } while (w == 0);
    if (stash - WREG)
        movwf stash ; *stash = w;
    endif
endm ;} // zOS_LAU()

zOS_ACT macro fsrnum
    local proceed,endact
    if (fsrnum & 3)
fsrn set 1
    else
fsrn set 0
    endif
    andlw 0x07 ;inline int zOS_ACT(uint8_t** fsrn, uint8_t w) {
    btfsc STATUS,Z ; if (w &= 0x07) { // activate valid job launch
    bra endact ;
    btfsc WREG,2 ; if (w < 6) { // prevent stomp on globals/SFRs
    btfss WREG,1 ; *fsrn = (w + 1) << 4; // structure for job w
    bra proceed ; (*fsrn)[zOS_PCH] &= 0x7f; // allowed to run
    bra endact ; } // else w was > 5
proceed

```

```

zos_MEM FSR#v(fsrn),WREG,zOS_PCH
movlw 0x7f ; } // else w was < 1
andwf INDF#v(fsrn),f ;
swapf FSR#v(fsrn)L,w ;
andlw 0x07 ; return w;
addlw 0xff ;}

endact
endm

zos_INI macro fsrnum,val0,vall
if (fsrnum & 3)
fsrn set 1
else
fsrn set 0
endif
;after: zOS_LAU FSR#v(fsrn)L
lslf FSR#v(fsrn)L,f ;inline void zOS_INI(uint8_t* fsrnum, uint8_t
movlw 0x70 ; val0, uint8_t vall) {
iorwf FSR#v(fsrn)L,f ; //fsrnum starts and ends as a launched job#
clrf FSR#v(fsrn)H ; fsrnum = 0x70 | (fsrnum << 1);
movlw val0 ; // change global mailbox to non-0 if desired
movwi FSR#v(fsrn)++ ; fsrnum[0] = val0;
movlw vall ;
movwi FSR#v(fsrn)-- ; fsrnum[1] = vall;
lsrf FSR#v(fsrn),w ; fsrnum = (fsrnum >> 1) & 0x07; // unchanged
andlw 0x07 ;}
endm

zos_DIS macro fsrnum,job ;inline void zOS_DIS(int8_t* *fsr, int8_t job) {
if (fsrnum & 3)
fsrn set 1
else
fsrn set 0
endif
if (job)
zos_MEM FSR#v(fsrn),job,zOS_HDH ; *fsr = 0x10 * (1+job) + zOS_HDH;//priv
btfsc INDF#v(fsrn),zOS_PRB ; if (**fsr & (1<<zOS_PRB))
endif
bcf INTCON,GIE ; INTCON &= ~(1<<GIE);
endm ;} // zOS_DIS()

zos_ENA macro ;inline void zOS_ENA(void) {
bsf INTCON,GIE ; INTCON |= 1<<GIE;
endm ;} // zOS_ENA()

zos_ARG macro arg
local num
num set (arg & 0x03)
if (num == 0)
bcf INTCON,GIE ;inline void zOS_ARG(const int8_t arg, int8_t w)
endif
movwf zOS_AR#v(num) ;{if (!arg) INTCON &= ~(1<<GIE); zOS_AR0[arg]=w;}
endm

zos_RUN macro t0enable,t0flags
;; start a TMR0 interrupt since none found (most in INTCON, others PIE0)
zos_T0E equ t0enable
zos_T0F equ t0flags
if (zos_T0E)
banksel zOS_T0E
bsf zOS_T0E,T0IE ;inline void zOS_RUN(uint8_t* t0enable) {
if (zos_T0E - INTCON)
bsf INTCON,PEIE ; if (t0enable) { *t0enable |= 1<<T0IE;
endif
endif
;; advance the stack pointer to allow 5 stacks of 3 each (+1 if running)
banksel STKPTR ; if (t0enable != INTCON) INTCON |= 1<<PEIE;
movlw zOS_BOS ; }

```

```

movwf STKPTR ; STKPTR = zOS_BOS; // every job bottom of stack

;; set the active job to the first (and potentially only), interrupts ON
movlw 1+zOS_NUM ; bsr_shad = w = 1+zOS_NUM; // will wrap around
movwf BSR_SHAD ; boot(); // run the scheduler to grab its PC
pagesel boot ;} // zOS_RUN()
call boot ;

boot
bsf INTCON,GIE ;void boot(void) { INTCON |= 1<<GIE; zOS_RFI();}
zos_RFI
endm

zos_DBG macro
local loop
banksel STKPTR
clrf STKPTR ;inline void zOS_DBG(void) {
clrw ; for (int8_t w = STKPTR = 0;

loop
clrf TOSH ; w < 16; w++){
movwf TOSL ; TOSH = 0;
incf STKPTR,w ; TOSL = w;
andlw 0x0f ;
movwf STKPTR ; STKPTR = (STKPTR + 1) % 16;
btfss STATUS,Z ; }
bra loop ; STKPTR = -1;
decf STKPTR,f ; // still in job "0"
movlb 0 ;} // zOS_DBG()
endm

#ifdef PID1CON
;; 16x16bit signed multiply zOS_AR1:0 * zOS_AR3:2, core yielded during 7ms math
zos_MUL macro fsrnum
local fn,inout,fac0L,fac0H,fac1L,fac1H,zeroH,start,con,setup,enb,bsy
if (fsrnum & 3)
fn set 1
else
fn set 0
endif
inout set 0x1f80 & PID1SETL
fac0L set 0x1f & PID1K1L
fac0H set 0x1f & PID1K1H
fac1L set 0x1f & PID1SETL
fac1H set 0x1f & PID1SETH
zeroH set 0x1f & PID1INH
start set 0x1f & PID1INL
con set 0x1f & PID1CON
out0 set 0x1f & PID1OUTLL
out1 set 0x1f & PID1OUTLH
out2 set 0x1f & PID1OUTHLL
out3 set 0x1f & PID1OUTHHL
setup set (1<<PID1MODE1)
enb set PID1EN
bsy set PID1BUSY

movlw low PID1CON ;void zOS_MUL(int16_t** fsr) {
movwf FSR#v(fn)L ; *fsr = &PID1CON;
movlw high PID1CON ;
movwf FSR#v(fn)H ; do {

spinget
btfss INDF#v(fn),enb ; while ((**fsr&(1<<enb))&& // MATHACC for sure
bra notbusy ; (**fsr&(1<<bsy))) // ours if not busy
btfss INDF#v(fn),bsy ; { // or never enabled
bra notbusy ; zOS_ARG(0, bsr);
movf zOS_ME ; zOS_SWI(zOS_YLD);
zos_ARG 0
zos_SWI zOS_YLD ; }
bra spinget ; // interrupts now enabled if zOS_SWI called

notbusy
bcf INTCON,GIE ; INTCON &= ~(1<<GIE);

```

```

        btfsc    INDF#v(fn),enb ; // begin critical section (seizing MATHACC)
        bra      spinget        ;
        bsf      INDF#v(fn),bsy ;
        bra      spinget        ; } while ((**fsr & (1<<enb)) || (**fsr & (1<<bsy)));
        movlw    setup          ;
        movwf    indf#v(fn)      ; **fsr = 1<<PIDMODEL; // unsigned mult no accum
        bsf      indf#v(fn),enb ; **fsr |= 1<<PID1EN; // selected, then enabled
        movlw    low inout      ;
        movwf    FSR#v(fn)L     ;
        movlw    high inout     ;
        movwf    FSR#v(fn)H     ; *fsr = &PID1SETL & 0x1f80; // just bank bits
        movf     zOS_AR3,w      ;
        movwi    fac0H[FSR#v(fn)]; (0x1f & PID1K1H)[*fsr] = zOS_AR3;
        movf     zOS_AR2,w      ;
        movwi    fac0L[FSR#v(fn)]; (0x1f & PID1K1L)[*fsr] = zOS_AR2;
        movf     zOS_AR1,w      ;
        movwi    fac1H[FSR#v(fn)]; (0x1f & PID1SETH)[*fsr] = zOS_AR1;
        movf     zOS_AR0,w      ;
        movwi    fac1L[FSR#v(fn)]; (0x1f & PID1SETL)[*fsr] = zOS_AR0;
        clrw     zeroH[FSR#v(fn)]; (0x1f & PID1INH)[*fsr] = 0;
        movwi    start[FSR#v(fn)]; // end critical section (seizing MATHACC)
        bsf      INTCON,GIE     ; INTCON |= 1<<GIE;
        movlw    low PID1CON    ;
        movwf    FSR#v(fn)L     ;
        movlw    high PID1CON   ; *fsr = &PID1CON;
        movwf    FSR#v(fn)H     ; do {
spinmul
#if 0
        clrw     clrwdt         ; clrwdt();
#endif
        movf     zOS_ME         ; zOS_ARG(0, bsr);
        zOS_ARG 0
        zOS_SWI zOS_YLD
        btfss    INDF#v(fn),bsy ; zOS_YLD();
        bra      spinmul        ; } while (**fsr & 1<<PID1BUSY);
        bcf      INTCON,GIE     ; INTCON &= ~(1<<GIE);
        bcf      INDF#v(fn),enb ; // begin critical section (copying result)
        movlw    low inout      ; **fsr &= ~(1<<enb); // disable MathACC to free
        movwf    FSR#v(fn)L     ;
        movlw    high inout     ;
        movwf    FSR#v(fn)H     ; *fsr = &PID1SETL & 0x1f80; // just bank bits
        moviw    out3[FSR#v(fn)]; zOS_AR3 = (0x1f & PID1OUTH)[*fsr];
        movwf    zOS_AR3        ;
        moviw    out2[FSR#v(fn)]; zOS_AR2 = (0x1f & PID1OUTHL)[*fsr];
        movwf    zOS_AR2        ;
        moviw    out1[FSR#v(fn)]; zOS_AR1 = (0x1f & PID1OUTLH)[*fsr];
        movwf    zOS_AR1        ;
        moviw    out0[FSR#v(fn)]; zOS_AR0 = (0x1f & PID1OUTLL)[*fsr];
        movwf    zOS_AR0        ; // end critical section (when ARx copy's done)
        bsf      INTCON,GIE     ; } // zOS_MUL()
        endm
#endif

zOS_PAG macro    fsrnum
        local    fsrn
        if (fsrnum & 3)
fsrn set 1
        else
fsrn set 0
        endif

        swapf    FSR#v(fsrn)L,w ;uint8_t zOS_PAG(void* fsrnum) {
        andlw    0x0f           ;
        bcf      FSR#v(fsrn)H,5 ;
        swapf    FSR#v(fsrn)H,f ;
        iorwf    FSR#v(fsrn)H,w ;
        swapf    FSR#v(fsrn)H,f ; return w = (fsrnum >> 4);
        bsf      FSR#v(fsrn)H,5 ;} // zOS_PAG()

```

```

        endm

zOS_PTR macro    fsrnum
        local    fsrn
        if (fsrnum & 3)
fsrn set 1
        else
fsrn set 0
        endif

        swapf    WREG,w         ;void zOS_PTR(void** fsrnum, uint8_t w) {
        movwf    FSR#v(fsrn)H   ;
        movwf    FSR#v(fsrn)L   ;
        movlw    0x0f           ;
        andwf    FSR#v(fsrn)H,f ;
        bsf      FSR#v(fsrn)H,4 ;
        movlw    0xf0           ; *fsrnum = 0x2000 | w<<4;
        andwf    FSR#v(fsrn)L,f ;} // zOS_PTR()
        endm

;;; must be defined with 2 SWI flags: one for malloc(), a different for free()
;;; (typically instantiated with base=0x2210, size = memory size - base)
;;; SWI behavior for malloc(w) is to return pointer in w of 2 middle nybbles
;;; in linear address space, e.g. 0x21 for first cell on a 5-job system, or 0
;;; in w if no free memory of size zOS_AR0*16 bytes was available
;;; SWI behavior for free(w) is to return in w the number of bytes now free/16
;;; intersecting with the address whose middle nybble is zOS_AR0, or 0 in w if
;;; zOS_AR0 didn't point to a valid (i.e. previously allocated) block of bytes

zOS_HEA macro    base,size,mi,fi ;void zOS_HEA(void* base, void* size, uint8_t
        local    isr,decl,task ; mi/*malloc*/,uint8_t fi/*free*/) {

        bra      decl           ; goto decl;

        local    maxnon0,allocated,always0,temp,adrrary,tblsize
        local    tblrows,sizarry,memroun,mem3nyb,membase,memsize
maxnon0 set      0x6c
allocated set    0x6d
always0 set      0x6e
temp set         0x6f
adrrary set      0x20
tblsize set      0x50
tblrows set      tblsize/2
sizarry set      adrrary+tblrows
memroun set      base+0xf
mem3nyb set      memroun&0xffff
membase set      mem3nyb>>4
memsize set      size>>4

isr
        local    mloop,mcandid,mexact,mnotall,groloop
        local    free,floop,ffound,invalid,done

        movf     zOS_JOB,w      ; isr:
        movwf    BSR            ; bsr = zOS_JOB;

        zOS_MY2 FSR1            ; fsr1 = 0x70|(bsr<<1);
        moviw    FSR1++         ;
        iorwf    INDF1,w        ;
        btfsc    STATUS,Z       ; if (0[fsr1] | 1[fsr1])
        bra      invalid        ; goto invalid; // not init'ed according to mbox

#if (mi - fi)
        movf     zOS_MSK,w      ;
        andlw    mi             ; //////////////////////////////////////////
        btfsc    STATUS,Z       ; //////////////////////////////////////////
        bra      free           ; if ((mi != fi) && (zOS_MSK & mi)) ||

#else
        movf     zOS_AR1,w      ; ((mi == fi) && (zOS_AR0!=*sic*/zOS_AR1)) {

```

```

        movf    zOS_AR0,f      ; // can either assign separate SWIs for malloc
        movwf   zOS_AR0       ; // and free or if nearing the SWI limit of 5,
        btfscc  STATUS,Z      ; // put the parameter in ARG1 instead of ARG0
        bra     free          ; // and ARG0!=0 for malloc() or ==0 for free()

#endif

        zOS_LOC FSR0,BSR,adrrary; for (fsr0 = (bsr<<7)+adrrary,
        zOS_LOC FSR1,BSR,sizarry; fsr1 = (bsr<<7)+sizarry;

mloop
        moviw   FSR0++        ; (allocated = temp = *fsr0++); // next poss.
        btfscc  STATUS,Z      ; fsr1++ {
        bra     invalid       ;
        movwf   temp          ;
        movwf   allocated     ;
        moviw   FSR1++        ; w = *fsr1++; // number of bytes used,0=freed
        btfscc  STATUS,Z      ;
        bra     mloop         ; if (w == 0) { // allocatable

mcandid
        moviw   0[FSR0]       ; w = *fsr0; // upper limit to allocating here
        btfscc  STATUS,Z      ; if (w == 0)
        bra     invalid       ; goto invalid; // past the highest address

        bsf     STATUS,C      ; // temp is now the address of this candidate
        comf    temp,f         ; // w is now the next address past candidate
        addwfc  temp,w         ;
        movwf   temp          ;
        subwf   zOS_AR0,w      ; else if ((w = zOS_AR0 - (temp = w-temp))>0)
        btfscc  STATUS,Z      ;
        bra     mexact        ; // -w now holds extra space beyond requested
        btfscc  WREG,7         ; // temp now holds total available at allocated
        bra     mloop         ;
        bra     mnottall       ; continue; // not enough allocatable here

mexact
        movf    zOS_AR0,w      ; if (w == 0) { // exactly enough!
        movwi   -1[FSR1]       ; -1[fsr1] = zOS_AR0; // allocated size
        moviw   -1[FSR0]       ; w = -1[fsr0]; // recycled handle
        bra     done           ; goto done;

mnottall
        movf    maxnon0,f      ; } else if (adrrary[tblrows-2] != 0) // full
        btfscc  STATUS,Z      ; goto invalid;
        bra     invalid       ;

        movf    zOS_AR0,w      ; // w == addr to insert, temp == size to insert
        movwi   -1[FSR1]       ; -1[fsr1] = zOS_AR0; // record it as granted
        clrf    temp           ; temp = 0;
        addwf   allocated,w     ; for (w = -1[fsr0] + temp; *fsr0; fsr0++,fsr1++

) {
groloop
        xorwf   INDF0,f        ; // w == contents for inserted cell for fsr0
        xorwf   INDF0,w        ; // *fsr0 == contents to overwrite in fsr0
        xorwf   INDF0,f        ; swap(&w, fsr0);

        xorwf   temp,f         ; // w == contents just overwritten in fsr0
        xorwf   temp,w         ; // temp == contents for inserted cell (fsr1)
        xorwf   temp,f         ; swap(&w, &temp);

        xorwf   INDF1,f        ; // w == contents for inserted cell in fsr1
        xorwf   INDF1,w        ; // *fsr1 == contents to overwrite in fsr1
        xorwf   INDF1,f        ; swap(&w, fsr1);

        xorwf   temp,f         ; // w == contents just overwritten in fsr1
        xorwf   temp,w         ; // temp == contents just overwritten in fsr0
        xorwf   temp,f         ; swap(&w, &temp);

        addfsr  FSR0,+1        ; // w == contents just overwritten in fsr0
        addfsr  FSR1,+1        ; // temp = contents just overwritten in fsr1

        movf    INDF0,f        ;
        btfscc  STATUS,Z      ;

```

```

        bra     groloop       ; }

        movwi   0[FSR0]       ; // append the final overwritten contents
        movf    temp,w        ; *fsr0 = w; // this will be maxnon0 for last
        movwi   0[FSR1]       ; *fsr1 = w = temp;
        movf    allocated,w    ; w = allocated;
        bra     done          ; goto done; // return the fsr0 address added

free
        movf    zOS_MSK,w      ; //////////////////////////////////////////
        andlw   fi             ; ////////////////////////////////// free() //
        btfscc  STATUS,Z      ;
        bra     invalid       ; } else if (zOS_MSK & fi)

        zOS_LOC FSR0,BSR,adrrary

floop
        moviw   FSR0++        ; for (fsr0 = (bsr<<7) + adrrary;
        xorwf   zOS_AR0,w      ; fsr0 < adrrary + tblrows; //FIXME:sorted!
        btfscc  STATUS,Z      ; fsr0++) //could quit early!
        bra     ffound        ;

        movlw   adrrary+tblrows ;
        xorwf   FSR0L,w        ;
        andlw   0x7f           ;
        btfscc  STATUS,Z      ;
        bra     floop         ;

        bra     invalid       ; if (*fsr0 == zOS_AR0) {

ffound
        if (tblrows & 0x20)
            addfsr FSR0,0x1f    ;
            addfsr FSR0,tblrows-0x1f;
        else
            addfsr FSR0,tblrows ; fsr0 = sizarry + (fsr0 - adrrary);
        endif
        moviw   --FSR0         ; w = *--fsr0;
        clrf    INDF0         ; *fsr0 = 0;
        bra     done          ; }

invalid
        clrw                    ; else invalid: w = 0; // can't malloc nor free

done
        zOS_RFS WREG          ; done: return w;

        zOS_NAM "heap allocator"
        zOS_NAM "malloc(),free(),garbage coll"

task
        local   iniarry,coalesec,coaloop,coscoot

        bcf     INTCON,GIE     ;task:
        zOS_LOC FSR0,BSR,0x70

iniarry
        clrw                    ; INTCON &= ~(1<<GIE);
        movwi   --FSR0         ; for (fsr0 = (bsr<<7)|(adrrary+tblsize);
        movlw   adrrary        ; fsr > adrrary; fsr--)
        xorwf   FSR0L,w        ; *fsr = 0; // zero each address and size entry
        andlw   0x7f           ;
        btfscc  STATUS,Z      ;
        bra     iniarry        ;

        zOS_MY2 FSR1

        movlw   membase        ; // except first address entry is start of heap
        movwi   0[FSR1]        ; (0x70|(bsr<<1))[0] =
        movwi   0[FSR0]        ; adrrary[0] = membase; // first allocatable
        movlw   membase+memsize ; // and second address entry is the end of heap
        movwi   1[FSR1]        ; (0x70|(bsr<<1))[1] =
        movwi   1[FSR0]        ; adrrary[1] = membase+memsize; //max allocatable

coalesec
        movf    zOS_ME        ; do { // combine adjacent rows whose size are 0

```

```

zos_ARG 0
zos_SWI zos_YLD
zos_LOC FSR0,BSR,adrrarry+1
zos_LOC FSR1,BSR,sizarry

coaloop
    bcf     INTCON,GIE      ; zos_ARG(0, bsr);
    moviw  ++FSR0          ; zos_SWI(zos_YLD); // only 1 pass per schedule
    btfsc  STATUS,Z        ; INTCON &= ~(1<<GIE); // critical section (
    bra    coalesc         ; for (fsr0 = &adrrarry[1], fsr1 = &sizarry[0];
    moviw  FSR1++          ;      **++fsr0; fsr1++)
    btfss  STATUS,Z        ; if (0[fsr1] == 0 && 1[fsr1] == 0) {
    bra    coaloop         ; INTCON |= 1<<GIE;
    moviw  0[FSR1]         ; do { // fsr1->redun row siz,trails fsr0->adr
    btfss  STATUS,Z        ; INTCON &= ~(1<<GIE); // critical section (
    bra    coaloop         ; uint8_t w = **++fsr1;

coscoot
    moviw  ++FSR1          ; -1[fsr1] = w;
    movwi  -1[FSR1]        ; w = *fsr0++;
    moviw  FSR0++          ; } while ((-2[fsr0] = w) != 0);
    movwi  -2[FSR0]        ; break; // ) critical section ended by SWI
    btfss  STATUS,Z        ; }
    bra    coscoot         ; } while (1);
    bra    coalesc         ; decl:

decl
    zos_ADR task,zos_UNP   ; fsr0 = task & 0x7fff; // MSB 0 => unprivileged
    movlw  low isr         ; w = zos_ARG(0, isr & 0x00ff);
    zos_ARG 0
    movlw  high isr        ; w = zos_ARG(1, isr>>8);
    zos_ARG 1
    movlw  0               ; w = zos_ARG(2, 0); // no hardware interrupts
    zos_ARG 2
    movlb  0               ; // still in job "0": don't forget this!!!!
    endm                  ; } // zos_HEA()

;;; simple output-only console job with circular buffer
zos_HEX macro
    andlw  0x0f            ;
    addlw  0x06            ;
    btfsc  WREG,4          ;inline char zos_HEX(uint8_t w) {
    addlw  0x07            ; return (w & 0x0f > 9) ? '0'+w : 'A'+w-10;
    addlw  0x2a            ; } // zos_HEX()
    endm

zos_IHF macro ofs,fsrsrc,fsrdst
    local src,dst
    if (fsrsrc & 3)
src set 1
    else
src set 0
    endif
    if (fsrdst & 3)
dst set 1
    else
dst set 0
    endif

    moviw  ofs[FSR#v(src)] ;inline void zos_IHF(int8_t ofs, int fsrnum,
    swapf  WREG,w          ; char* file) {
    zos_HEX
    movwi  FSR#v(dst)++    ; file[0] = zos_HEX(ofs[fsrnum] >> 4);
    moviw  ofs[FSR#v(src)] ; file[1] = zos_HEX(ofs[fsrnum]);
    zos_HEX
    movwi  FSR#v(dst)++    ; } // zos_IHF()
    endm

zos_UNW macro job          ;inline void zos_UNW(int8_t job) { }
    zos_MEM FSR0,job,zos_PCH; fsr0 = 0x10 * (1 + job) + zos_PCH;

```

```

    bcf     INDF0,zos_WAI   ; *fsr0 &= ~(1<< zos_WAI); // now runnable
    endm                  ; } // zos_UNW()

zos_OUT macro swinum,str,temp
    local agent,pre,post,setup,len,sloop,loop
    bra     setup           ;inline void zos_OUT(uint8_t swinum, char* str,
agent      brw             ; uint8_t* temp) { // no '\0'
    pre     dt              str
    post    set             post-pre
    len     if (len > 254)
            error "string too long"
    endif

    if (len)
setup      movlw len        ; zos_SWI(zos_YLD); // get buffer empty as poss.
            movwf temp      ; for (*temp = strlen(str); *temp; --*temp) {
sloop      movf  zos_ME      ;
            zos_ARG 0
            zos_SWI zos_YLD
loop       movf  temp,w      ; zos_ARG(0, w = str[strlen(str) - *temp]);
            sublw len        ; while (zos_SWI(swinum) != 1) { // buffer full
pagesel    agent
            call agent       ; zos_SWI(zos_YLD); // flush buffer, retry
            zos_ARG 0
    else
sloop      movf  zos_ME      ;
            zos_ARG 0
            zos_SWI zos_YLD
setup      if (temp - zos_AR0)
            if (temp - WREG)
                movf temp,w      ;
            endif
            zos_ARG 0
            endif
            endif

zos_SWI    swinum
decfsz    WREG              ; zos_ARG(0, w = str[strlen(str) - *temp]);
bra       sloop            ; }

    if (len)
        decfsz temp,f        ; }
        bra     loop         ; } // zos_OUT()
    endif
endm

zos_PSH macro reg
    movf  zos_ME              ;inline void zos_PSH(uint8_t* reg) {
    ; bcf INTCON,GIE
    banksel TOSH
    incf  STKPTR,f           ; STKPTR++; // caller should've masked interrupts
    movwf TOSH               ; TOSH = bsr; // must store bsr so we can go back
    if (reg=BSR)
        movf reg,w          ; if (reg != &bsr)
        movwf TOSL          ; TOSL = *reg;
        movf TOSH,w         ; bsr = TOSH;
    endif
    movwf BSR                ; } // zos_PSH()
    ; bsf INTCON,GIE
endm

```

```

zos_POP macro reg
;; bcf INTCON,GIE
banksel STKPTR
if (reg=BSR)
    movf TOSL,w ;inline void zos_POP(uint8_t* reg) {
    movwf reg ; if (reg != &bsr) *reg = TOSL;
endif
movf TOSH,w ; bsr = TOSH;
decf STKPTR,f ; STKPTR--;// caller should've masked interrupts
movwf BSR ;} // zos_POP()
;; bsf INTCON,GIE
endm

zos_RDF macro
#ifdef EADRL
zos_ADL equ EADRL
zos_ADH equ EADRH
zos_RDL equ EEDATL
zos_RDH equ EEDATH
banksel EECON1
bcf EECON1,CFGs ;inline void zos_RDF(void) { // for EADR micros
bsf EECON1,EEPGD ; EECON1 &= ~(1<<CFGs);
bsf EECON1,RD ; EECON1 |= 1<<EEPGD;
nop ; EECON1 |= 1<<RD;
nop ;} // zos_RDF()
#else
#ifdef PMADRL
zos_ADL equ PMADRL
zos_ADH equ PMADRH
zos_RDL equ PMDATL
zos_RDH equ PMDATH
banksel PMCON1
bcf PMCON1,CFGs ;inline void zos_RDF(void) { // for PMADR micros
bsf PMCON1,RD ; PMCON1 &= ~(1<<CFGs);
nop ; PMCON1 |= 1<<RD;
nop ;} // zos_RDF()
#else
#ifdef NVMADRL
zos_ADL equ NVMADRL
zos_ADH equ NVMADRH
zos_RDL equ NVMDATL
zos_RDH equ NVM DATH
banksel NVMCON1
bcf NVMCON1,NVMREGS ;inline void zos_RDF(void) { // for NVM micros
bsf NVMCON1,RD ; NVMCON1 &= ~(1<<CFGs); NVMCON1 |= 1<<RD;
#endif
#endif
#endif
endm ;} // zos_RDF()

zos_STR macro swinum
local loop,done
bcf INTCON,GIE ;inline void zos_STR(const char* fsr0,
zos_PSH BSR ;
banksel zos_ADL ; uint8_t swinum) {
movf FSR0L,w ;
movwf zos_ADL ; INTCON &= ~(1<<GIE);
movf FSR0H,w ; zos_PSH(&bsr); // need a bank change for reads
movwf zos_ADH ; for (zos_AD = fsr0; *zos_AD; zos_AD++) {
loop
    zos_RDF
    rlf zos_RDL,w ; zos_RDF(); // read packed 14-bit contents
    rlf zos_RDH,w ;
    btfs STATUS,Z ;
    bra done ; if ((w = (zos_RDH<<1)|(zos_RDL>>7)) != '\0'){
    movwf zos_AR0 ; zos_ARG(0, w);
    zos_POP BSR
    zos_OUT swinum,"",zos_AR0

```

```

    bcf INTCON,GIE ; zos_POP(&bsr); // back to the expected bank
    zos_PSH BSR
    banksel zos_RDL
    movf zos_RDL,w ; zos_OUT(swinum,"",zos_AR0); // print ASCII
    andlw 0x7f ; INTCON &= ~(1<<GIE); // undo SWI GIE toggle
    btfs STATUS,Z ; zos_PSH(&bsr);
    bra done ; if ((w = zos_RDL & 0x7f) != '\0') {
    movwf zos_AR0 ; zos_ARG(0, w);
    zos_POP BSR
    zos_OUT swinum,"",zos_AR0
    bcf INTCON,GIE ; zos_POP(&bsr); // back to the expected bank
    zos_PSH BSR
    banksel zos_ADL
    incfsz zos_ADL,f ; zos_SWI(swinum,"",zos_AR0); // print ASCII
    bra loop ; INTCON &= ~(1<<GIE); // undo SWI GIE toggle
    incf zos_ADH,f ; zos_PSH(&bsr);
    bra loop ; } else break;
done
    zos_POP BSR ; } else break;
    bsf INTCON,GIE ; } zos_POP(&bsr); INTCON |= 1<<GIE;
    endm ;} // zos_STR()

zos_PUT macro fsrnum,max,wrap,p
local fsrn
if (fsrnum & 3)
    fsrn set 1
else
    fsrn set 0
endif
movwi FSR#v(fsrn)++ ;inline int8_t zos_PUT(char**fsrnum,uint7_t max,
movf FSR#v(fsrn)L,w ; char* wrap, char* p, char w) {
andlw 0x7f ; *(&fsrnum)++ = w;
xorlw max ; // w gets put in buffer regardless, but caller
swapf wrap,w ; // only updates the local pointer if not full
btfs STATUS,Z ; // (i.e. Z not set) by xor return value with p
swapf FSR#v(fsrn)L,w ; *fsrnum = (*fsrnum&0x7f==max) ? wrap : *fsrnum;
swapf WREG ; return (*fsrnum & 0x00ff) ^ p; //0 if full, or
movwf FSR#v(fsrn)L ; // new pointer value xor p if not
xorwf p,w ;} // zos_PUT()
endm

zos_BUF macro fsrnum,max,ptr
local ascii,err1,done
local fsrn
if (fsrnum & 3)
    fsrn set 1
else
    fsrn set 0
endif
lsrf zos_ME ;inline int8_t zos_BUF(char**fsrnum,uint7_t max,
movwf FSR#v(fsrn)H ; char** ptr, char w) { // p0, p1, wrap
movf 1+ptr,w ; // must be in job bank already, interrupts off
movwf FSR#v(fsrn)L ; fsr0 = (bsr<<7) | ptr[1]; // insertion pointer

movf zos_AR0,w ; if ((w = zos_AR0) == 0) { // 2-digit hex byte
btfs STATUS,Z ; w = zos_HEX(zos_AR1>>4); // convert high nyb
bra ascii ; w = zos_PUT(fsrnum, max, ptr[0], w); // room?

swapf zos_AR1,w ; if (w == 0)
zos_HEX
zos_PUT fsrnum,max,2+ptr,ptr
btfs STATUS,Z ; return 0; // buffer was full
bra done ; ptr[1] = w^ptr[0]; // correctly updated
xorwf ptr,w ; w = zos_HEX(zos_AR1); // convert low nybble
movwf 1+ptr ; w = zos_PUT(fsrnum, max, ptr[0], w); // room?

movf zos_AR1,w ; if (w == 0)
zos_HEX
zos_PUT fsrnum,max,2+ptr,ptr

```

```

        btfsc STATUS,Z      ; return 1; // buffer filled after first char
        bra  err1           ; ptr[1] = w^ptr[0]; // correctly updated
        xorwf ptr,w         ; w = 2;
        movwf 1+ptr         ; } else { // print an ascii character
        movlw 2             ; if ((w = zOS_PUT(fsrnum,max,ptr[0],w)) == 0)
        bra  done           ; return 0; // buffer was full

ascii
        zOS_PUT fsrnum,max,2+ptr,ptr
        btfsc STATUS,Z      ; ptr[1] = w^ptr[0]; // correctly updated
        bra  done           ; w = 1;
        xorwf ptr,w         ; }
        movwf 1+ptr         ; return w; // num of characters added to buffer

err1
        movlw 1             ;} // zOS_BUF()

done
        endm

zOS_NUL macro hwflag        ;void zOS_NUL(void) { // replacement for zOS_CON
        bra  decl           ; goto decl;
        local task,isr,decl ; task: do {

task
        movf zOS_ME         ; zOS_ARG(0, bsr);
        zOS_ARG 0
        zOS_SWI zOS_YLD     ; zOS_SWI(zOS_YLD);
        bra  task           ; } while (1);

isr
        banksel zOS_T0F     ; isr:
        bcf zOS_T0F,T0IF    ; zOS_T0F &= ~(1<<T0IF); // clear interrupt flag
        zOS_RFI             ; zOS_RFI(); // and go back to scheduler

decl
        zOS_ADR task,zOS_UNP ; fsr0 = task & 0x7fff; // MSB 0 => unprivileged
        movlw low isr       ; w = zOS_ARG(0, isr & 0x00ff);
        zOS_ARG 0
        movlw high isr      ; w = zOS_ARG(1, isr>>8);
        zOS_ARG 1           ; w = zOS_ARG(2, 1<<T0IF);
        movlw hwflag        ; w = zOS_ARG(3, 0 /* no SWI */);
        zOS_ARG 2
        clrw                ;} // zOS_NUL()
        zOS_ARG 3
        movlb 0             ; // still in job "0": don't forget this!!!!
        endm

zOS_CON macro p,rat,rts,hb,pin;inline void zOS_CON(int8_t p,int8_t rat,int8_t
        local contask,conisr,initd,conloop,condecl
        bra  condecl        ; rts,int8_t* hb,int8_t pin){

        ;; initialize constants and variables
        local t0div,t0rst
        t0div set 0
        t0rst set 1

        local p0,p1,wrap,t0scale,isradrl,isradrh,tskadrl,tskadrh,optadrl
        local optadrh,accumul,accumuh,numbase,destreg,destreh,char_io,buf,max

        ;; 0x20~24 reserved for zOS_CON
        p0 set 0x20
        p1 set 0x21
        wrap set 0x22
        t0scale set 0x23

        ;; 0x24~28 reserved for zOS_INP
        isradrl set 0x24
        isradrh set 0x25
        tskadrl set 0x26
        tskadrh set 0x27

        ;; 0x28~2F reserved for zOS_MON and derivations e.g. zOS_MAN

```

```

optadrl set 0x28
optadrh set 0x29
accumul set 0x2a
accumuh set 0x2b
numbase set 0x2c
destreg set 0x2d
destreh set 0x2e
char_io set 0x2f
buf set 0x30
max set 0x70

```

;copy the preceding lines rather than including this file, as definitions for
;zOS_MON()-derived macros referring to these local variables wouldn't open it
;until expansion and would throw an undefined-var error during the processing

```

        local uatbase,uatxmit
        if (p == 0)
        uatbase set TXREG & 0xff80
        uatxmit set TXREG & 0x001f ; mask off just the SFR space
        rtsflag set TXIF
        else
        uatbase set TX#v(p)REG & 0xff80
        uatxmit set TX#v(p)REG & 0x001f ; mask off just the sfr SFR
        rtsflag set TX#v(p)IF
        endif
        zOS_NAM "console (output-only)"

contask
        movlw high uatbase ; goto decl;
        movwf FSR0H        ;task:// all init that requires knowledge of BSR
        zOS_MY2 FSR0
        moviw t0div[FSR0] ; do {
        btfss STATUS,Z    ; fsr0 = (uatbase & 0xff00) | 0x0070 |(bsr<<1);
        bra  initd        ; if (1[fsr0] == 0) { // not initialized yet
        zOS_DIS GIE,0
        movlw 0xff        ; zOS_DIS(&fsr0, zOS_JOB); // interrupts off!
        movwi t0div[FSR0] ; 0[fsr0] = 0xff; // live TMR0 postscaler divider
        movlw 0x00        ;
        movwi t0rst[FSR0] ; 1[fsr0] = 0x00; // live reset value for TMR0
        rrf zOS_ME        ;
        clrw                ; const char* max = 0x70;
        rrf WREG           ; static char *p0, *p1, buf[]; //p0:task, p1:ISR
        iorlw buf          ; const char* wrap = ((bsr&1)<<7) | buf;
        movwf wrap        ; p0 = p1 = wrap; // reset value if they max out
        movwf p0           ; zOS_ENA(); // interrupts on after init done
        movwf p1           ; puts("\r\nWelcome to zOS\r\n");
        zOS_ENA ;//FIXME: superfluous due to subsequent SWI
        zOS_OUT 0xff,"\r\nWelcome to zOS\r\n",char_io

initd
        movf zOS_ME        ; zOS_ARG(0, bsr);
        zOS_ARG 0
        zOS_SWI zOS_YLD     ;
        movlw low uatbase ; const int8_t* uatbase = uatxmit & 0xff80;
        movwf FSR0L        ; fsr0 = uatbase;
        movlw high rts     ; zOS_ARG(0, bsr);
        movwf FSR1H        ; zOS_SWI(zOS_YLD);
        movlw low rts      ; // wait for SWI to store char(s) in buf[]
        movwf FSR1L        ;
        btfss INDF1,rtsflag ; if (*(fsr1 = rts) & (1<<rtsflag) == 0) //full
        bra  conloop       ; continue; //yield (still sending or no char)
        lsrz zOS_ME        ;
        movwf FSR1H        ; // READY TO SEND, AND...
        zOS_DIS GIE,0
        movf p0,w          ; // begin critical section (freeze pointers)
        movwf FSR1L        ;
        xorwf p1,w         ; fsr1 = (bsr<<7) | p0;
        btfsc STATUS,Z    ; if (p0 == p1)
        bra  conloop       ; continue; // nothing to do
        moviw FSR1++        ;
        movwi uatxmit[FSR0] ; uatxmit[fsr0] = *fsr1++; // send a character

```



```

    movf    FSR1L,w      ;
    movwf   p0           ; p0 = fsr1 & 0x00ff; // wrap around to buf+0
    andlw   0x7f         ;
    xorlw   max          ;
    btfss   STATUS,Z     ;
    bra     conloop      ; if (p0 & 0x7f == max) // ignore low bank bit
    movf    wrap,w       ; p0 = wrap; // =buf xor the lowest bank bit
    movwf   p0           ; // end critical section
conloop
    zOS_ENA
    zOS_MEM FSR0,BSR,0
    moviw   zOS_HDH[FSR0] ;
    movwf   PCLATH       ;
    moviw   zOS_HDL[FSR0] ;
    movwf   PCL          ; } while (1); // e.g. might run zOS_INP's task

    ;; HWI will be coming from a tmr0 expiration, for the blinking heartbeat
    ;;
    ;; SWI will be coming from a job that wants to send a character
    ;; in which case the ISR stores it, advancing pl and returning the
    ;; number of characters stored in the buffer
    ;; Note: caller needs to make sure to check status of return value for
    ;; != 0, just in case job is in between sleeps or with a full buffer
conisr
    local   done,do_swi,nottmr

    ;; if it's a simple and frequent timer overflow interrupt finish quickly
    bankssel zOS_T0F
    btfss   zOS_T0F,T0IF ; if (/*presumed true:(zOS_T0E & (1<<T0IE)) &&*/
    bra     nottmr       ; (zOS_T0F & (1<<T0IF))) { // timer overflow
    bcf     zOS_T0F,T0IF ; zOS_T0F &= ~(1<<T0IF); // clear interrupt flag

    ;; get fsr0 pointing to tmr0 postscaler/reset value
    movf    zOS_JOB,w     ; isr:
    movwf   BSR           ; bsr = zos_job;
    zOS_MY2 FSR0L         ; fsr0 = 0x70 | (bsr < 1);

    ;; with fsr0 pointing to global pair, point fsr1 to local mem("t0scale")
    zOS_LOC FSR1,zOS_JOB,t0scale
    bankssel TMR0
    moviw   t0rst[FSR0]   ; fsr1 = (zOS_JOB << 7) | t0scale;
    btfss   WREG,7        ; bsr = TMR0 >> 7; //now invalid for this branch
    movwf   TMR0          ; if (t0rst[fsr0] < 128) // max 7 bit TMR0 reset
    decfsz  INDF1,f       ; TMR0 = t0rst[fsr0]; // or chance of deadlock
    bra     done          ; if (--*fsr1 == 0) {

    bankssel hb
    movf    INDF0,w       ;
    btfsc   STATUS,Z      ;
    movlw   1             ; if (*fsr0 == 0) // disallow zero postscaler
    movwf   INDF0         ; *fsr0 = 1;
    movwf   INDF1         ; *fsr1 /*countdown*/ = *fsr0 /*postscaler*/;
    movlw   (1<<pin)      ;
    xorwf   hb,f          ; hb ^= 1 << pin;
    bra     done          ; } else {

    ;; check for validated SWI first since it will be in zOS_MSK, else a HWI
nottmr
    movf    zOS_MSK,f     ; if (zOS_MSK) { // a SWI to buffer a character
    btfss   STATUS,Z      ; w = zOS_BUF(&fsr0, max, p0); // zOS_AR0,_AR1
    bra     do_swi        ; zOS_RFS(w); } else zOS_RET(); // not ours(!)
    zOS_RET

    ;; point fsr0 to uatbase (again?), point fsr1 to p0
do_swi
    movf    zOS_JOB,w     ;
    movwf   BSR           ;
    zOS_BUF FSR0,max,p0   ; }
    zOS_RFS WREG          ; zOS_RFI(); // HWI finished

```

```

done
    zOS_RFI              ;

    ;; initialize the UART peripheral, job handle and first three arguments
condecl
    bankssel uatbase
    bcf     RCSTA,SPEN    ; decl: // all init that is BSR independent here
    bcf     RCSTA,CREN    ; RCSTA &= ~(1<<SPEN)|(1<<CREN));
    bcf     TXSTA,TXEN    ; TXSTA &= ~(1<<TXEN);
    local   brgval,brgvalm,brgvalh,brgvall
#ifdef BRG16
    brgval  set           rat>>2
    brgvalm set           brgval-1
    brgvalh set           high brgvalm
    brgvall set           low brgvalm
    bsf     BAUDCON,BRG16 ; // section 26.1.2.8 of 16F1847 steps below:
#endif SYNC
    bcf     TXSTA,SYNC    ; // (1) "Initialize..the desired baud rate"
#else
    bcf     TXSTA,SYNC_TXSTA
#endif
    bsf     TXSTA,BRGH    ; BAUDCON |= 1<<BRG16; // 16-bit generator
    movlw   brgvall       ; TXSTA &= ~(1<<SYNC); // async mode
    movwf   SPBRGL        ; TXSTA |= 1<<BRGH; // high speed
    movlw   brgvalh       ;
    movwf   SPBRGH        ; SPBRG = (rat/4) - 1;
    bcf     BAUDCON,SCKP  ; BAUDCON &= ~(1<<SCKP); // "SCKP..if inverted"
#else
    brgval  set           rat>>4
    brgvalm set           brgval-1
    brgvalh set           0
    brgvall set           low brgvalm
    bsf     TXSTA,BRGH    ; TXSTA |= 1<<BRGH; // (1) the desired baud rate
    movlw   brgvall      ;
    movwf   SPBRG         ; SPBRG = (rat/16) - 1;
#endif
    bsf     RCSTA,SPEN    ; // (3) "Enable..by setting..SPEN"
    bcf     RCSTA,RX9     ; RCSTA &= ~(1<<RX9); // (5) "9-bit..set..RX9"
    bsf     RCSTA,CREN    ; RCSTA |= (1<<SPEN) | (1<<CREN); // (6) "CREN"
    bsf     TXSTA,TXEN    ; TXSTA |= 1<<TXEN; // (5) "Enable..by..TXEN"
    bankssel PIE1
    bsf     PIE1,RCIE     ; PIE1 |= 1<<RCIE; //(4) "Set..RCIE..and..PEIE"
    zOS_ADR contask,zOS_PRB ; fsr0 = contask & 0x7fff; // MSB 1 => privileged
    movlw   low conisr    ; w = zOS_ARG(0, conisr & 0x00ff);
    zOS_ARG 0
    movlw   high conisr   ; w = zOS_ARG(1, conisr>>8);
    zOS_ARG 1             ; w = zOS_ARG(2, (0<<TXIF)|(1<<T0IF));
    movlw   (0<<TXIF)|(1<<T0IF)
    zOS_ARG 2
    movlb   0             ; // still in job "0": don't forget this!!!!
    endm                 ; } // zOS_CON()

    ;; like zOS_CON, but also accepts console input for command-line interaction
zOS_INP macro p,ra,rt,h,pi,isr;inline void zOS_INP(int8_t p, int8_t ra, int8_t
    local   rxtask,no_opt,rxisr,rxdecl
    bra     rxdecl        ; rt, int8_t* h, int8_t pi, void(*isr)()) {

    ;; reserve constants and variables
    local   p0,pl,wrap,t0scale,isradrl,isradrh,tskadrl,tskadrh,optadrl
    local   optadrh,accumul,accumuh,numbase,destreg,destreh,char_io,buf,max

    ;; 0x20~24 reserved for zOS_CON
    p0      set           0x20
    pl       set           0x21
    wrap     set           0x22
    t0scale  set           0x23

    ;; 0x24~28 reserved for zOS_INP
    isradrl set           0x24

```

```
isradrh set 0x25
tskadrl set 0x26
tskadrh set 0x27
```

```
;; 0x28~2F reserved for zOS_MON and derivations e.g. zOS_MAN
```

```
optadrl set 0x28
optadrh set 0x29
accumul set 0x2a
accumuh set 0x2b
numbase set 0x2c
destreg set 0x2d
destreh set 0x2e
char_io set 0x2f
buf set 0x30
max set 0x70
```

```
;copy the preceding lines rather than including this file, as definitions for
;zOS_MON()-derived macros referring to these local variables wouldn't open it
;until expansion and would throw an undefined-var error during the processing
```

```
local uarbase,uarecv,rxflag
if (p == 0)
uarbase set RCREG & 0xff80
uarecv set RCREG & 0x7f
rxflag set RCIF
else
uarbase set RC#v(p)REG & 0xff80
uarecv set RC#v(p)REG & 0x7f
rxflag set RC#v(p)IF
endif
```

```
zOS_NAM "console I/O"
```

```
;;; FIXME: haven't actually written the var init code for zOS_MON et al yet
```

```
rxtask
movf optadrh,w ; goto rxdecl;
movwf PCLATH ; rxtask:
iorwf optadrl,w ;
btfsc STATUS,Z ;
bra no_opt ;
movf optadrl,w ; if ((optadrh<<8) | optadrl)
callw ; (*(optadrh<<8) | optadrl) (); //returns to:
;;; FIXME: do anything interesting with return value? 0 sent if nothing happened
no_opt
movf tskadrh,w ;
movwf PCLATH ; goto (tskadrh<<8) | tskadrl; // zOS_CON() code
movf tskadrl,w ;
movwf PCL ; callw ; // will retrieve its own address as a loop
```

```
rxisr
movf zOS_JOB,w ; rxisr:
movwf BSR ; bsr = zOS_JOB; // isr starts with unknown bank

movf isradrh,w ;
movwf PCLATH ;
movf isradrl,w ; if (rt && (1<<RCIF) == 0) // SWI, not inp char
banksel rt
btfss rt,rxflag ; goto (isradrh<<8)|isradrl; //zOS_CON takes SWI
movwf PCL ; else {
bcf rt,rxflag ; rt &= ~(1<<RCIF);

#ifdef CAUTIOUS
btfss RCSTA,OERR ;
bra noovrrn ; if ((uarbase | RCSTA) & (1<<OERR)) {
movlw '!' ; zOS_AR0 = '!';
movwf zOS_AR0 ; zOS_BUF(zOS_JOB, p0);
zOS_BUF FSR0,max,p0 ; }
noovrrn
#endif
banksel uarbase
movf uarecv,w ; // this read removes it from the FIFO
```

```
#ifdef CAUTIOUS
btfss RCSTA,OERR ; if (RCSTA & (1<<OERR)) // rx overrun
bcf RCSTA,CREN ; RCSTA &= ~(1<<CREN); // cleared by disable
bsf RCSTA,CREN ; RCSTA |= 1<<CREN; // (re-)enable reception
#endif

if (isr)
movwf zOS_AR0 ; zOS_AR0 = RCREG;
pagesel isr ; if (zOS_AR0)
btfss STATUS,Z ; goto isr; // continue with parser
goto isr ; zOS_RFI(); //return from interrupt
endif
zOS_RFI ; }
```

```
local vars,arg0,arg1,adrl,adrh,optl,opth,accl,acch,base,dstl,dsth,chio
vars
set 0x20
arg0 set isradrl-vars
arg1 set isradrh-vars
adrl set tskadrl-vars
adrh set tskadrh-vars
optl set optadrl-vars
opth set optadrh-vars
accl set accumul-vars
acch set accumuh-vars
base set numbase-vars
dstl set destreg-vars
dsth set destreh-vars
chio set char_io-vars
```

```
rxdecl
zOS_CON p,ra,rt,h,pi
zOS_LAU zOS_JOB
zOS_ACT FSR1
zOS_LOC FSR1L,zOS_JOB,vars
movf zOS_AR0,w ; rxdecl:
movwi arg0[FSR1] ; zOS_CON(p,ra,rt,h,pi); // extend zOS_CON()
movf zOS_AR1,w ; zOS_LAU(&fsr1); // by rewriting after launch
movwi arg1[FSR1] ; fsr1 <= 7;
movf FSR0L,w ; isradrl[fsr1] = (zOS_AR1<<8) | zOS_AR0;
movwi adrl[FSR1] ;
movf FSR0H,w ;
movwi adrh[FSR1] ; tskadr[fsr1] = fsr0; // still zOS_CON's handle
movlw 0 ;
movwi optl[FSR1] ; // caller sets optional task
movwi opth[FSR1] ; optadr[fsr1] = ((*void)()) 0; // no func
movwi accl[FSR1] ;
movwi acch[FSR1] ;
movwi dstl[FSR1] ;
movwi dsth[FSR1] ;
movwi chio[FSR1] ; char_io[fsr1] = 0; // zero = no action to take
movlw 0x0a ;
movwi base[FSR1] ;
rlf FSR1L,w ; w = fsr1 >> 7; // restore zOS_LAU() job number
rlf FSR1H,w ;
zOS_MEM FSR0,WREG,0
movlw low rxtask ; fsr0 = 0x10 + w << 4;
movwi zOS_HDL[FSR0] ;
movwi zOS_PCL[FSR0] ;
movlw high rxtask ;
movwi zOS_PCH[FSR0] ; zOS_PC[fsr0] = rxtask;
iorlw 0x80 ;
movwi zOS_HDH[FSR0] ; zOS_HD[fsr0] = rxtask | 0x8000;
addfsr FSR0,zOS_ISR ; fsr0 += zOS_ISR; // last 4 bytes of job record
movlw low rxisr ; *fsr0++ = rxisr & 0x00ff;
movwi FSR0++ ;
movlw high rxisr ; *fsr0++ = rxisr >> 8;
movwi FSR0++ ;
movf zOS_AR2,w ; *fsr0++ |= (1<<RCIF); // |(0<<TXIF)|(1<<T0IF);
iorlw 1<<rxflag ; // still in job "0"; caller sets any SWI value
movwi FSR0++ ; // zOS_INP()
```

```

endm

zOS_ACC macro valregs,basereg
    clrf valregs ;inline uint8_t zOS_ACC(uint8_t* valregs,uint8_t
    clrf 1+valregs ; *basereg) { // w unclobbered
    clrf basereg ; *valregs = 0;
    bsf basereg,3 ; return *basereg = 10; // decimal by default
    bsf basereg,1 ;} // zOS_ACC()
endm

zOS_PCT macro reg
    movlw 0x7e ; // 0 <= reg <= 100
    andwf reg,w ; w = reg & 0x7e; // 0 <= w <= reg (even, trunc)
    lslf reg,f ;
    lslf reg,f ; uint16_t c = reg * 4; // 0 <= reg <= 400
    btfs STATUS,C ; if (c > 0xff)
    iorlw 0x01 ; w |= 1;
    addwf reg,f ; c = reg += w;
    btfs STATUS,C ; if (c > 0xff)
    iorlw 0x01 ; w |= 1;
    rrf WREG ; // 0 <= (w&1)*256 + reg <= 500
    rrf reg,f ; reg = ((w&1)*256 + reg)/2; // 0 <= reg <= 250
endm

zOS_SEL macro adr0,adr1,file,b
    addlw low adr0 ;inline int zOS_SEL(char* adr0, char* adr1,
    clrf FSR0L ; uint8_t file, uint3_t b,
    addwfc FSR0L,f ; uint8_t w, char** fsr0) {
    movlw adr1 - adr0 ;
    btfs file,b ;
    addwfc FSR0L,f ;
    movlw high adr0 ;
    movwf FSR0H ; fsr0 = w + ((file & (1<<b)) ? adr1 : adr0);
    clrw ; return 0;
    addwfc FSR0H,f ;}
endm

zOS_DEC macro putch,puts,enc,retadr

    local ophi_0X,ophi_11,bitops,literal,onelit,litbyte,calllit,bradest
    local destreg,onedest,nametst,namereg,flagreg,regarg2,endopc
    local overld0,nodest,overld1,overld2,braneg,brapos,overld3,omnibus
    local noargs,newbank,moviw,moveoffs,nameoffs
    local offset0,offset1,minfsr,minmin,plufsr,pluplu,opc_miw,opc_mwi
    local opc_lit,opc_mlp,opc_af0,opc_af1,opc_reg,opc_mov,opc_bit,opccall
    local opcgoto,opcclrw,opc_bpo,opc_bng,opcomni,opc_mlb,hexpref
    local regnam0,regnam1,regnam2,regnam3,regnam4,regnam5
    local regnam6,regnam7,regnam8,regnam9,regnamA,regnamB

    movlw 0x1f ;void zOS_DEC(uint14_t enc) {
    andwf 1+enc,w ; uint8_t w = (enc &= 0x1fff) >> 8;
    btfs 1+enc,5 ;
    bra ophi_0X ;
    btfs 1+enc,4 ;
    bra calllit ; if ((enc & 0x3000 == 0x3000) ||
    bra ophi_11 ; (enc & 0x3000 == 0)) { // not b_/call/goto

    ophi_0X
    btfs 1+enc,4 ;
    bra bitops ; enc = w; // builds string index in bits 8~12

    ophi_11
    clrf 1+enc ; switch (w) { case 0: /*
    brw ;movwf/callw/movlb/brw/retfie/return/clrwdt/nop/
    bra overld0 ;option/reset/sleep/tris/mov[w]*/ goto overld0;
    bra overld1 ;/* 0x01nn=>clrf/clrw*/ case 1: goto overld1;
    bra destreg-0x12 ;/* 0x02nn => subwf */ case 2: goto destreg-18;
    bra destreg-0x11 ;/* 0x03nn => decf */ case 3: goto destreg-17;
    bra destreg-0x10 ;/* 0x04nn => iorwf */ case 4: goto destreg-16;

    bra destreg-0xf ;/* 0x05nn => andwf */ case 5: goto destreg-15;
    bra destreg-0xe ;/* 0x06nn => xorwf */ case 6: goto destreg-14;
    bra destreg-0xd ;/* 0x07nn => addwf */ case 7: goto destreg-13;
    bra destreg-0xc ;/* 0x08nn => movf */ case 8: goto destreg-12;
    bra destreg-0xb ;/* 0x09nn => comf */ case 9: goto destreg-11;
    bra destreg-0xa ;/* 0x0ann => incf */ case 10: goto destreg-10;
    bra destreg-9 ;/* 0x0bnn => decfsz */ case 11: goto destreg-9;
    bra destreg-8 ;/* 0x0cnn => rrf */ case 12: goto destreg-8;
    bra destreg-7 ;/* 0x0dnn => rlf */ case 13: goto destreg-7;
    bra destreg-6 ;/* 0x0enn => swapf */ case 14: goto destreg-6;
    bra destreg-5 ;/* 0x0fnn => incfsz */ case 15: goto destreg-5;

    bra literal-6 ;/* 0x30nn => movlw */ case 16: goto literal-6;
    bra overld2 ;/* 0x31nn movlp/addfsr */ case 17: goto overld2;
    bra brapos ;/* 0x32nn => bra(fwd) */ case 18: goto brapos;
    bra braneg ;/* 0x33nn => bra(rev) */ case 19: goto braneg;
    bra literal-5 ;/* 0x34nn => retlw */ case 20: goto literal-5;
    bra destreg-4 ;/* 0x35nn => lslf */ case 21: goto destreg-4;
    bra destreg-3 ;/* 0x36nn => lsrw */ case 22: goto destreg-3;
    bra destreg-2 ;/* 0x37nn => asrf */ case 23: goto destreg-2;
    bra literal-4 ;/* 0x38nn => iorlw */ case 24: goto literal-4;
    bra literal-3 ;/* 0x39nn => andlw */ case 25: goto literal-3;
    bra literal-2 ;/* 0x3ann => xorlw */ case 26: goto literal-2;
    bra destreg-1 ;/* 0x3bnn => subwfb */ case 27: goto destreg-1;
    bra literal-1 ;/* 0x3cnn => sublw */ case 28: goto literal-1;
    bra destreg-0 ;/* 0x3dnn => addwfc */ case 29: goto destreg-0;
    bra literal-0 ;/* 0x3enn => addlw */ case 30: goto literal-0;
    bra overld3 ;/* 0x3fnn movwi/iw [] */ case 31: goto overld3;

    bitops
    andlw 0x0c ; } else if (enc & 0x3000 == 0x1000) { // bit op
    addlw low opc_bit ;// fortuitously, opcodes are separated by 4 in
    movwf FSR0L ;// enc as well as the opcode strings of 4 words
    movlw high opc_bit ;
    movwf FSR0H ;
    clrw ;
    addwfc FSR0H,f ;
    pagesel puts
    call puts ; puts(fsr0 = bit_lit[w /*0,4,8 or 12*/ >>2]);
    movlw 0x03 ;
    andwf 1+enc,f ; enc[1] &= 0x03; // bit number < 8
    rlf enc,w ; enc[1] <= 1; // pull in bit 7 from low byte:
    rlf 1+enc,f ; enc[1] |= (w & 0x80) ? 1 : 0; // bit number<8
    lslf 1+enc,f ; enc[1] <= 1; // bit number now in bits 3:1
    bsf 1+enc,0 ; enc[1] |= 1; // and now C is set for puts
    bra nametst ; goto nametst; // handle known register names

    incf 1+enc,f ; // opc_lit[6] = "movlw 0"
    incf 1+enc,f ; // opc_lit[5] = "retlw 0"
    incf 1+enc,f ; // opc_lit[4] = "iorlw 0"
    incf 1+enc,f ; // opc_lit[3] = "andlw 0"
    incf 1+enc,f ; // opc_lit[2] = "xorlw 0"
    incf 1+enc,f ; // opc_lit[1] = "sublw 0"

    literal
    lslf 1+enc,w ; } literal: // opc_lit[0] = "addlw 0"
    lslf WREG ;
    addlw low opc_lit ;
    movwf FSR0L ;
    movlw high opc_lit ;
    movwf FSR0H ;
    clrw ;
    addwfc FSR0H,f ; fsr0 = opc_lit[w];
    movlw 0xff ; w = 0xff;

    onelit
    andwf enc,f ;onelit:
    pagesel puts
    call puts ; enc &= w;
    movf enc,f ; puts(fsr0);
    zOS_ADR hexpref,zOS_FLA

```

```

        pagesel puts
        call      puts          ; puts("0x");
litbyte movf      enc,w          ; litbyte:
        pagesel putch
        bsf      STATUS,C        ; putch(enc & 0xff, c = 1); // as hexadecimal
        call     putch           ; return;
        bra      endopc          ; }
calllit movlw     low opccall     ;
        bcf      STATUS,C        ;
        btfsc    1+enc,3         ;
        addlw    opcgoto-opccall ;
        movwf    FSR0L           ;
        movlw    high opccall    ;
        movwf    FSR0H           ;
        clrw     ;
        addwfc   FSR0H,f         ;
        pagesel puts
        call     puts           ; puts(fsr0 = opccall[w /*0 or 4*/ >> 2];
        movlw    0x07           ;
bradest andwf     1+enc,w         ;
        pagesel putch
        bsf      STATUS,C        ;
        call     putch           ; putch((enc&0x700) >> 8,c=1); // as hexadecimal
        bra      litbyte        ; goto litbyte; // lsb above, to save space

        incf     1+enc,f         ; // opc_reg[18] = "subwf "
        incf     1+enc,f         ; // opc_reg[17] = "decf "
        incf     1+enc,f         ; // opc_reg[16] = "iorwf "
        incf     1+enc,f         ; // opc_reg[15] = "andwf "
        incf     1+enc,f         ; // opc_reg[14] = "xorwf "
        incf     1+enc,f         ; // opc_reg[13] = "addwf "
        incf     1+enc,f         ; // opc_reg[12] = "movf "
        incf     1+enc,f         ; // opc_reg[11] = "comf "
        incf     1+enc,f         ; // opc_reg[10] = "incf "
        incf     1+enc,f         ; // opc_reg[9] = "decfsz "
        incf     1+enc,f         ; // opc_reg[8] = "rrf "
        incf     1+enc,f         ; // opc_reg[7] = "rlf "
        incf     1+enc,f         ; // opc_reg[6] = "swapf "
        incf     1+enc,f         ; // opc_reg[5] = "incfsz "
        incf     1+enc,f         ; // opc_reg[4] = "lsrf "
        incf     1+enc,f         ; // opc_reg[3] = "lsrf "
        incf     1+enc,f         ; // opc_reg[2] = "asrf "
        incf     1+enc,f         ; // opc_reg[1] = "subwfb "
destreg lslf      1+enc,w         ; // opc_reg[0] = "addwfc "
        clrf     1+enc           ; //so test between w and f will happen for wf's
        lslf     WREG            ;
        addlw    low opc_reg     ;
        movwf    FSR0L           ;
        movlw    high opc_reg    ;//FIXME: needs comments
onedest movwf     FSR0H           ;
        clrw     ;
        addwfc   FSR0H,f         ; // carry set by jumper!!
        pagesel puts
        call     puts           ;
nametst movf      enc,w          ;
        andlw    0x7f            ;
        addlw    0-0x0c          ;
        btfsc    WREG,7          ;
        bra      namereg        ;
        zOS_ADR hexpref,zOS_FLA ;
        pagesel puts
        call     puts           ;
        movf     enc,w           ;

        andlw    0x7f            ;
        pagesel putch
        bsf      STATUS,C        ;
        call     putch           ;
        bra      flagreg        ;
namereg movf      enc,w          ;
        andlw    0x0f            ;
        pagesel nameoff
        call     nameoff         ;
        addlw    low regnam0     ;
        movwf    FSR0L           ;
        movlw    high regnam0    ;
        movwf    FSR0H           ;
        clrw     ;
        addwfc   FSR0H,f         ;
        pagesel puts
        call     puts           ;
flagreg incf      1+enc,w         ;
        btfsc    STATUS,Z        ; if (enc & 0xff00 == 0xff00)
        bra      endopc          ; return;
        movlw    ','            ;
        pagesel putch
        bcf      STATUS,C        ;
        call     putch           ;
        lsr     1+enc,w         ;
        btfsc    STATUS,C        ;
        bra      regarg2        ;
        movlw    'f'            ;
        btfsc    enc,7           ;
        movlw    'w'            ;
regarg2 pagesel putch
        call     putch           ;
endopc  pagesel retadr
        goto     retadr         ;
overld0 movlw     0xff           ;
        movwf    1+enc           ; enc |= 0xff00; // special, allows: bra onedest
        movlw    low opc_mov     ;
        movwf    FSR0L           ;
        movlw    high opc_mov    ;
        bcf      STATUS,C        ;
        btfsc    enc,7           ;
        bra      onedest        ;
        bra      omnibus        ;
onedest movwf     FSR0H           ;
        clrw     ;
        addwfc   FSR0H,f         ;
        pagesel puts
        call     puts           ;
        pagesel retadr
        goto     retadr         ;
overld1 movlw     low opcclrw     ;
        bcf      STATUS,C        ;
        btfsc    enc,7           ;
        addlw    4               ; // carry handled in onedest
        movwf    FSR0L           ;
        movlw    0xff            ;
        movwf    1+enc           ;
        movlw    high opcclrw    ;
        btfsc    enc,7           ;
        bra      onedest        ;
        bra      nodest         ;
overld2

```

```

    movlw    low opc_mlp    ;
    movwf    FSR0L         ;
    movlw    high opc_mlp   ;
    movwf    FSR0H         ;'
    movlw    0x7f          ;
    btfsc    enc,7          ;
    bra      onelit        ;
    movlw    0              ;
    btfsc    enc,6          ;
    movlw    opc_af1-opc_af0 ;
    addlw    low opc_af0    ;
    movwf    FSR0L         ;
    movlw    high opc_af0   ;
    movwf    FSR0H         ;
    clrw     ;
    addwfc   FSR0H,f        ;
    movlw    0x1f          ;
    btfss    enc,5          ;
    bra      onelit        ;
    movlw    0xc0          ;
    iorwf    enc,f          ;
    movlw    0xff          ;
    bra      onelit        ;

#if 0
braneg
    comf     enc,f          ;
    incf     enc,f          ; enc = -enc;
    movlw    opc_bng-opc_bpo ;
    bra      brapos+1       ;

brapos
    movlw    0              ;
    addlw    low opc_bpo    ;
    movwf    FSR0L         ;
    movlw    high opc_bpo   ;
    movwf    FSR0H         ;
    clrw     ;
    addwfc   FSR0H,f        ;
    movlw    0xff          ;
    bra      onelit        ;

opc_bpo
da      "bra  +"

opc_bng
da      "bra  -"

#else
braneg
    movlw    0xff          ;
    movwf    1+enc          ;

brapos
    movf     3+enc,w        ;
    addwfc   enc,f          ; // the caller already updated
    movf     4+enc,w        ;
    addwfc   1+enc,f        ;
    movlw    low opc_bra    ;
    movwf    FSR0L         ;
    movlw    high opc_bra   ;
    movwf    FSR0H         ;
    pagesel  puts           ;
    call     puts           ;
    movlw    0x7f          ;
    bra      bradest        ;

opc_bra
da      "bra    0x"

#endif

overld3
    swapf    enc,w          ; w = enc >> 4;
    pagesel  putmovi        ;
    call     putmovi        ;
    zOS_ADR  zero,zOS_FLA

```

```

    movf     enc,w          ; fsr0 = "0";
    andlw    0x3f           ;
    movwf    1+enc          ; enc[1] = enc[0] & 0x3f; // enc keeps FSRn's n
    btfsc    STATUS,Z       ;
    bra      printfn        ; if (enc[1] != 0) {
    zOS_ADR  hexpref,zOS_FLA
    btfss    enc,5          ; fsr0 = "0x";
    bra      printfn        ; if ((int6_t)(enc[1]) < 0) {
    movlw    0xc0           ;
    iorwf    1+enc,f        ; enc[1] = (int6_t)(enc[1]); // sign-extend
    comf     1+enc,f        ;
    incf     1+enc,f        ; enc[1] = -(enc[1]);
    movlw    2              ;
    addwfc   FSR0L,f        ;
    clrw     ;
    addwfc   FSR0H,f        ; fsr0 = "-0x";
    printfn  ; }

    pagesel  puts           ;
    call     puts           ; puts(fsr0);
    pagesel  putch          ;
    movf     1+enc,w        ;
    bsf      STATUS,C       ;
    call     putch          ; putch(enc[1], c = 1); // hexadecimal value

    printfn  ;
    swapf    enc,w          ; }
    andlw    0x04           ;
    addlw    low offset0    ;
    movwf    FSR0L         ;
    movlw    high offset0   ;
    movwf    FSR0H         ;
    clrw     ;
    addwfc   FSR0H,f        ; fsr0 = (enc & 0x40) ? "[FSR1]" : "[FSR0]";
    pagesel  puts           ;
    call     puts           ; puts(fsr0);
    pagesel  retadr         ; return;
    goto     retadr         ;}

newbank
    movlw    low opc_mlb    ;
    movwf    FSR0L         ;
    movlw    high opc_mlb   ;
    movwf    FSR0H         ;
    movlw    0x1f          ;
    bra      onelit        ;

omnibus
    btfsc    enc,6          ; // we know bit 7 (movwf) is clear
    bra      noargs         ;
    movlw    0xf0           ;
    andwf    enc,w          ;
    btfsc    STATUS,Z       ;
    bra      noargs         ; // 0x0_ and 0x6_ are arg-less
    btfsc    enc,5          ;
    bra      newbank        ;

    movf     enc,w          ; // movwi/iw with auto pre/post incr/decrement
    pagesel  putmovi        ;
    call     putmovi        ; putmovi(enc); // bit3 0/1 => moviw/movwi
    movlw    0x04           ;
    andwf    enc,w          ; w = enc & 0x04; // 0 for FSR0, 4 for FSR1
    btfsc    enc,1          ;
    bra      post           ; if (enc & (1<<1) == 0) // pre incr/decrement
    zOS_SEL  fsrprei,fsrpred,enc,0
    pagesel  puts           ;
    call     puts           ; puts(fsr0);
    pagesel  retadr         ;
    goto     retadr         ; else // post incr/decrement

post
    zOS_SEL  fsrposi,fsrposd,enc,0

```

```

    pagesel puts
    call puts ; puts(fsr0);
    pagesel retadr
    goto retadr ; }

noargs
    lslf enc,w ;
    lslf WREG ;
    andlw 0x3c ; w = (enc & 0x0f) * 4; // uniform string length
    addlw low opcomni ;
    movwf FSR0L ;
    movlw high opcomni ;
    bra nodest ;

putmovi
    lsrif WREG ;void putmovi(uint4_t w) {
    andlw 0x04 ;
    addlw low opc_miw ;
    movwf FSR0L ;
    movlw high opc_miw ;
    movwf FSR0H ;
    clrw ; // moviw/wi distinction always at nybble top
    addwfc FSR0H,f ; puts((w & (1<<4)) ? "movwi " : "moviw ");
    pagesel puts
    goto puts ;}

zero
    da "0"

offset0
    da "[FSR0]",0

offset1
    da "[FSR1]",0

fsrprei
    da "++FSR0",0
    da "++FSR1",0

fsrpred
    da "--FSR0",0
    da "--FSR1",0

fsrposi
    da "FSR0++",0
    da "FSR1++",0

fsrposd
    da "FSR0--",0
    da "FSR1--",0

opc_miw
    da "moviw "

opc_mwi
    da "movwi "

opc_lit
    da "addlw "
    da "sublw "
    da "xorlw "
    da "andlw "
    da "iorlw "
    da "retlw "
    da "movlw "

opc_mlb
    da "movlb "

opc_mlp
    da "movlp "

opc_af0
    da "addfsr FSR0",0

opc_af1
    da "addfsr FSR1",0

opc_reg
    da "addwfc "
    da "subwfb "

```

```

    da "asrf "
    da "lsrf "
    da "lslf "
    da "incfsz "
    da "swapf "
    da "rlf "
    da "rrf "
    da "decfsz "
    da "incf "
    da "comf "
    da "movf "
    da "addwf "
    da "xorwf "
    da "andwf "
    da "iorwf "
    da "decf "
    da "subwf "

opc_mov
    da "movwf "

opc_bit
    da "bcf "
    da "bsf "
    da "btfsc "
    da "btfss "

opccall
    da "call 0x"

opcgoto
    da "goto 0x"

opcclrw
    da "clrw "
    da "clrf "

opccomni
    da "nop "
    da "reset "
    da "option "
    da "sleep "
    da "clrwdt "
    da "tris A "
    da "tris B "
    da "tris C "
    da "return "
    da "retfie "
    da "callw "
    da "brw "
    da "invalid"
    da "invalid"
    da "invalid"
    da "invalid"

hexpref
    da "0x",0

hex_neg
    da "-0x"

regnam0
    da "INDF0"

regnam1
    da "INDF1"

regnam2
    da "PCL"

regnam3
    da "STATUS",0

regnam4
    da "FSR0L"

regnam5
    da "FSR0H"

```

```

regnam6      da      "FSR1L"
regnam7      da      "FSR1H"
regnam8      da      "BSR"
regnam9      da      "WREG",0
regnamA      da      "PCLATH",0
regnamB      da      "INTCON",0
nameoff
    brw        ;
    retlw      regnam0-regnam0 ;
    retlw      regnam1-regnam0 ;
    retlw      regnam2-regnam0 ;
    retlw      regnam3-regnam0 ;
    retlw      regnam4-regnam0 ;
    retlw      regnam5-regnam0 ;
    retlw      regnam6-regnam0 ;
    retlw      regnam7-regnam0 ;
    retlw      regnam8-regnam0 ;
    retlw      regnam9-regnam0 ;
    retlw      regnamA-regnam0 ;
    retlw      regnamB-regnam0 ;
    endm

zOS_MON macro    p,ra,rt,h,pi,isr;inline void zOS_MON(int8_t p, int8_t ra, int8_t
    local    endmon
    pagesel  endmon          ;      rt, int8_t* h, int8_t pi, void(*isr)()) {
    goto     endmon          ; zOS_INP(p,ra,rt,h,pi,monisr); }// isr may be 0

    local    monisr,monchr1,monchr2,monchr3,mondump,mondest,monram,monchr4
    local    monchr5,monchr6,monchr7,monchr8,monchr9,monprmp,monlast,monpctg

    local    p0,p1,wrap,t0scale,isradrl,isradrh,tskadrl,tskadrh,optadrl
    local    optadrh,accumul,accumuh,numbase,destreg,destreh,char_io,buf,max

    ; 0x20~24 reserved for zOS_CON
    p0      set      0x20
    p1      set      0x21
    wrap    set      0x22
    t0scale set      0x23

    ; 0x24~28 reserved for zOS_INP
    isradrl set      0x24
    isradrh set      0x25
    tskadrl set      0x26
    tskadrh set      0x27

    ; 0x28~2F reserved for zOS_MON and derivations e.g. zOS_MAN
    optadrl set      0x28
    optadrh set      0x29
    accumul set      0x2a
    accumuh set      0x2b
    numbase set      0x2c
    destreg set      0x2d
    destreh set      0x2e
    char_io set      0x2f
    buf      set      0x30
    max      set      0x70

```

;copy the preceding lines rather than including this file, as definitions for
;zOS_MON()-derived macros referring to these local variables wouldn't open it
;until expansion and would throw an undefined-var error during the processing

```
#ifdef zOS_MIN
```

```

#else
    local    monout,loop,done,disasmb,disasmr,monpack

monpack
#ifdef CAUTIOUS
    movf     BSR,w
    movwf    zOS_JOB
#endif

    lsrwf    zOS_JOB,w          ;void monpack(char w, uint14t* fsr0) {
    movwf    FSR1H              ; // zos_job = bsr;
    movf     1+p0,w             ;
    movwf    FSR1L              ; fsr1 = (zos_job<<7) | ptr[1];
    banksel  zOS_ADL
    movf     FSR0L,w            ; // switches banks; GIE must be clear
    movwf    zOS_ADL            ; zOS_ADL = fsr0 & 0x00ff;
    movf     FSR0H,w            ; zOS_ADH = fsr0 >> 8;
    movwf    zOS_ADH            ; while (1) {

loop
    zOS_RDF
    rlf      zOS_RDL,w          ; zOS_RDF(); // read packed 14-bit contents
    movwf    FSR0L              ; //1st char:
    rlf      zOS_RDH,w          ; fsr0h = (zOS_RDH<<1) | ((zOS_RDL&0x80)?1:0);
    movwf    FSR0H              ; //2nd char:
    lsrwf    FSR0L,f            ; fsr0l = zOS_RDL & 0x7f;
    movf     zOS_JOB,w          ;
    movwf    BSR                ; bsr = zos_job; // back in buffer's bank

    movf     FSR0H,w            ; if ((w = fsr0h) == 0)
    btfsc    STATUS,Z           ; break; // null terminator in high byte
    bra      done               ; zOS_PUT(&fsr1, max, ptr[0], w);
    zOS_PUT  FSR1,max,2+p0,p0
    btfsc    STATUS,Z           ;//FIXME: pasted from zOS_BUF(), needs comments
    bra      done               ;"
    xorwf    p0,w               ;"
    movwf    1+p0               ;"

    movf     FSR0L,w            ; if ((w = fsr0l) == 0)
    btfsc    STATUS,Z           ; break; // null terminator in low byte
    bra      done               ; zOS_PUT(&fsr1, max, ptr[0], w);
    zOS_PUT  FSR1,max,2+p0,p0
    btfsc    STATUS,Z           ;//FIXME: pasted from zOS_BUF(), needs comments
    bra      done               ;"
    xorwf    p0,w               ;"
    movwf    1+p0               ;"

    banksel  zOS_ADL
    incfsz   zOS_ADL,f          ; bsr = zOS_ADL>>7; // back in flash-read bank
    bra      loop              ; if ((zOS_ADL = (zOS_ADL + 1) & 0x00ff) == 0)
    incf     zOS_ADH,f          ; zOS_ADH++;
    bra      loop              ; }

done
    return                          ;}

monout
    pagesel  monbufs
    btfss    STATUS,C           ;void monout(char w, uint1_t c) { // zOS_DEC arg
    goto     monbufs            ; if (c == 0) monbufs(w); else monlsb(w);
    pagesel  monlsb
    goto     monlsb            ;}

disasmb
    movlw    ' '                ;
    pagesel  monbufs
    call     monbufs            ;
    zOS_DEC  monout,monpack,accumul,disasmr

#endif

monback
    andlw    0x3f                ;void monback(uint3_t job, uint8_t ptr, char w){

```

```

        btfsc STATUS,Z      ; if (w &= 0x3f) {
        return              ; // 63 \b's should be enough in a buffer of 64
        movwf zOS_AR1      ;
#if 0
monbac2
        movf p0,w           ; // don't actually want to wind back buffer;
        xorwf pl,w          ; // the point is show what will be overwritten
        btfsc STATUS,Z      ;
        bra monbarn         ;
        movf pl,w           ;
        xorwf wrap,w        ;
        movlw max-1         ;
        btfss STATUS,Z      ;
        movwf pl            ;
        btfsc wrap,7        ;
        bsf pl,7            ;
        decf pl,f           ;
        decfsz zOS_AR1,f    ;
        bra monbac2         ;
        return              ;
monbarn
#endif
        movlw 0x08          ;
        movwf zOS_AR0       ; zOS_AR0 = '\b'; // FIXME: or '\0177'?

monloop
        zOS_BUF FSR0,max,p0
        andlw 0x1           ; for (zOS_AR1 = w; zOS_AR1; zOS_AR1--) {
        btfsc STATUS,Z      ; if (zOS_BUF(job, ptr) == 0) // buff full
        return              ; return;
        decfsz zOS_AR1,f    ; }
        bra monloop         ; }
        return              ; // monback() monloop()

monhex
        movf accumuh,w      ;void monhex(void) { monlsb(,w = accumuh); }
monlsb
        clrf zOS_AR0         ;void monlsb(uint3_t job, uint8_t ptr, char w) {
        movwf zOS_AR1        ;
        zOS_BUF FSR1,max,p0  ; return zOS_BUF(&fsr,ptr,w); } // 0/1/2 printed
        return              ; // monlsb

mon0
        movlw '0'           ;void mon0(void) { zOS_AR0 = '0'; monbufs(ptr);
        bra monbufs         ;}

monx
        movlw 'x'           ;void monx(void) { zOS_AR0 = '0'; monbufs(ptr);
        bra monbufs         ;}

monspc
        movlw ' '           ;void monspc(void) { zOS_AR0 = ' '; monbufs(ptr);
        bra monbufs         ;}

monlf
        movlw '\n'          ; return zOS_BUF(zos_job, ptr, w);

monbufs
        movwf zOS_AR0       ;} // monCrLf() monlf()

monbufd
        movlw 1             ;void monbufs(uint8_t ptr, char w) {
        movwf zOS_AR1        ; goto monloop();
        bra monloop         ;} //FIXME: these comments above are useless

monisr
        movf zOS_JOB,w       ;void monisr(void) {
        movwf BSR            ; bsr = zos_job;// to access char_io var et al
        pagesel monbufd      ;
        movlw 0xe0           ; // from zOS_INP isr with char zOS_AR0>0
        addwf zOS_AR0,w      ;
        btfss WREG,7         ; // refuse to echo unprintable characters

        call monbufd         ; if (zOS_AR0 > 31 && monbuf(zos_job,p0) > 0) {
        andlw 0x1            ; // successful echo into circular buffer
        pagesel monlast      ;
        btfsc STATUS,Z       ;
        goto monlast         ;

        movf zOS_AR0,w       ; // handle '~' before the tolower() conversion
        xorlw '~'            ;
        btfss STATUS,Z       ;
        bra monchr1          ; if (zOS_AR0 == '~') {
        pagesel mon0          ;
        call mon0             ;
        pagesel monx          ;
        call monx             ;
        comf accumul,f        ; accumul = ~accumul;
        comf accumuh,w        ;
        movwf accumuh         ;
        movwf char_io         ; char_io = accumuh = ~accumuh; // preserve
        pagesel monhex        ;
        call monhex           ; monhex(zos_job, p0);
        movf accumul,w        ; accumuh = accumul; // accumuh overwritten
        movwf accumuh         ; monlsb(zos_job, p0);
        pagesel monlsb        ;
        call monlsb           ; accumuh = char_io; // accumuh now restored
        movf char_io,w        ; char_io = 0; // completely handled in ISR
        movwf accumuh         ; zOS_RFI();
        clrf char_io          ; }
        zOS_RFI

monchr1
        btfsc zOS_AR0,6      ; if (zOS_AR0 & 0x40)
        bcf zOS_AR0,5        ; zOS_AR0 &= 0xdf; // zOS_AR0=tolower(zOS_AR0)
        movf zOS_AR0,w       ; //FIXME: ' { | } ~ DEL mapped onto @ [ \ ] ^ _
        movwf char_io        ;
        xorlw 0x08           ; switch (char_io = zOS_AR0) {
        movlw 0x7f           ;
        btfss STATUS,Z       ; case '\b':
        movf char_io,w        ;
        xorlw 0x7f           ;
        btfss STATUS,Z       ; case '\0177':
        bra monchr2          ;
        movlw '\r'           ;
        pagesel monbufs       ;
        call monbufs          ; monbuf(zos_job, p0, '\r');
        bra monprmp          ; goto monprmp;

monchr2
        movf char_io,w        ;
#if 0
        xorlw 0x0a           ;
        movlw 0x0d           ;
        btfss STATUS,Z       ; case '\n':
        movf char_io,w        ;
#endif
        xorlw 0x0d           ;
        btfss STATUS,Z       ; case '\r':
        bra monchr3          ; monbuf(zos_job, p0, '\n');// follows the \r
        movlw '\r'           ;
        pagesel monbufs       ;
        call monbufs          ;
        pagesel monlf         ;
        call monlf           ;

        movf destreg,w        ; // repeat \r's can set a whole range of
        movwf FSR0L           ; // addresses to zero???
        movf 1+destreg,w      ;
        movwf FSR0H           ; fsr0 = destreg;
        iorwf FSR0L,w         ;
        btfsc STATUS,Z       ;

```



```

bra    monprmp      ;   if (fsr0) { // destreg was set by ' ' or =
movf   accumul,w    ;   if (fsr0 & 0x8000 == 0)
btfss  FSR0H,7      ;
movwi  FSR0++,      ;   *fsr0 = accumul & 0x00ff; // not in flash
movf   FSR0L,w      ;
movwf  destreg      ;
movf   FSR0H,w      ;   destreg++; // advances for next access
movwf  l+destreg    ;   }
bra    monprmp      ;   goto monprmp;

monchr3
movf   char_io,w    ;
xorlw  ' ',         ;
movlw  ' ',         ;
btfsc  STATUS,Z     ;
movwf  char_io      ;   case ' ': // synonym for ' '
movf   char_io,w    ;
xorlw  ' ',         ;
btfsc  STATUS,Z     ;   case ' ':
bra    mondump      ;
movf   char_io,w    ;
xorlw  ' ',         ;
btfsc  STATUS,Z     ;   case '.':
bra    mondump      ;
movf   char_io,w    ;
xorlw  '=',         ;
btfss  STATUS,Z     ;   case '=':
bra    monchr4      ;

mondump
movf   accumul,w    ;   // pressing ' ' or '.' or '=' should apply
iorwf  accumul,w    ;   // to the recently incremented address from
btfsc  STATUS,Z     ;   // a previous operation (if any) or to an
bra    mondest      ;   // an address typed immediately before it
movf   accumul,w    ;
movwf  destreg      ;
movf   accumul,w    ;   if (accumul) // typed a value before ' ' /=
movwf  l+destreg    ;   destreg = accumul; // otherwise no clobber
movf   char_io,w    ;   if (char_io == ' ') {
xorlw  ' ',         ;   char_io = 0; // all we do is a destreg xfer
btfsc  STATUS,Z     ;   break;
bra    monzero      ;   }

mondest
btfss  l+destreg,7  ;   if (destreg & 0x8000) { // flash, not RAM
bra    monram       ;
pagesel mon0        ;
call   mon0         ;   putchar('0');
pagesel monx        ;
call   monx         ;   putchar('x');
movf   destreg,w    ;
movwf  FSR0L        ;
movf   l+destreg,w  ;
movwf  FSR0H        ;   fsr0 = destreg;
zos_PSH BSR
banksel zOS_ADL     ;
movf   FSR0L,w      ;   zOS_PSH(&bsr);
movwf  zOS_ADL      ;
movf   FSR0H,w      ;
movwf  zOS_ADH      ;   zOS_AD = fsr0;
zos_RDF
movf   zOS_RDH,w    ;   zOS_RDF();
movwf  zOS_AR0      ;   zOS_ARG(0,zOS_RDH); // only way to access
zos_POP BSR
movf   zOS_AR0,w    ;   zOS_POP(&bsr);
movwf  accumul      ;
pagesel monhex      ;
call   monhex       ;   monhex(zos_job, p0, accumul=0); // high byte
movf   destreg,w    ;
movwf  FSR0L        ;

movf   l+destreg,w  ;
movwf  FSR0H        ;   fsr0 = destreg; // monhex() clobbered fsr0
movwi  FSR0++,      ;
movwf  accumul      ;
movf   FSR0L,w      ;
movwf  destreg      ;   accumul = *fsr0++;
movf   FSR0H,w      ;   destreg = fsr0;
movwf  l+destreg    ;   monlsb(zos_job, p0, accumul); //   LSB
movf   accumul,w    ;
pagesel monlsb      ;
call   monlsb       ;   moncrlf(zos_job, p0); //   \r\n

#ifdef zOS_MIN
#else
local  disasmb,disasmr
pagesel disasmb
goto  disasmb      ;   goto disasmb; disasmr:
disasmr
#endif
movlw  '\r'
pagesel monbufs
call   monbufs
pagesel monlf
call   monlf        ;   goto monprmp;
bra    monprmp      ;   }

monram
pagesel mon0
call   mon0         ;
pagesel monx
call   monx         ;
movf   destreg,w    ;
movwf  FSR0L        ;
movf   l+destreg,w  ;
movwf  FSR0H        ;   fsr0 = destreg;
movwi  FSR0++,      ;
movwf  accumul      ;   accumul = *fsr0++;
pagesel monhex
call   monhex       ;   monhex(p0, accumul);

movf   char_io,w    ;
xorlw  ' ',         ;   // then exits in the '.' case to just print
btfss  STATUS,Z     ;   if (char_io == '.') {
bra    monramd      ;
movf   FSR0L,w      ;
movwf  destreg      ;
movf   FSR0H,w      ;
movwf  l+destreg    ;   destreg = fsr0;
movlw  '\r'         ;   monbufs('\r');
pagesel monbufs
call   monbufs      ;   monbufs('\n');
pagesel monlf
call   monlf        ;   goto monprmp;
bra    monprmp      ;   }

monramd
movf   char_io,w    ;   // or follow by 3 backspaces in the '=' case
xorlw  ' ',         ;   // to show that \r will result in a 0 write
btfss  STATUS,Z     ;
movlw  3            ;
pagesel monback
call   monback      ;   monback(zos_job, p0, (char_io == '.')?0:3);
clrf  char_io       ;   char_io = 0;
zos_RFI char_io     ;   break;

monchr4
movf   char_io,w    ;
xorlw  'X'          ;
btfss  STATUS,Z     ;   case 'X':
bra    monchr5      ;
movlw  0x10         ;   numbase = 16;

```

```

movwf  numbase      ; char_io = 0;
clrf   char_io      ; break;
zos_RFI

monchr5
movf   char_io,w    ;
xorlw  '%'          ;
btfss  STATUS,Z     ; case '%':
bra    monchr6      ;
movlw  0x9b         ;
addwf  accumul,w    ;
btfsc  WREG,7       ;
bra    monpctg      ; if (accumul > 102)
movlw  0x66         ;
movwf  accumul      ; accumul = 102;

monpctg
movf   accumul,w    ; accumul = zOS_PCT(accumul);
zos_PCT accumul
movf   accumul,w    ; monecho:
movwf  accumuh      ; accumuh = accumul;
pagesel monhex      ; monhex(zos_job, p0); print as e.g. 50%0x7d
call   monhex       ; accumuh = 0;
clrf   accumuh      ; char_io = 0;
clrf   char_io      ; break;
zos_RFI

monchr6
movlw  0-0x30       ; default:
addwf  char_io,f    ;
btfsc  char_io,7    ;
bra    monchr9      ; if ((char_io -= ('0'&0xdf /*0x10*/)) >= 0) {
movlw  0-0x10       ;
addwf  char_io,w    ;
btfsc  WREG,7       ; if (char_io > 0x10)
bra    $+3          ;
movlw  0xf9         ;
addwf  char_io,f    ; char_io -= 0x07; // 0x41->0x11->0x0a... so
btfss  STATUS,Z     ; // or :=0x0a,...,?=0x0f,
bra    monchr7      ; // or A=0x2a,B=0x2b,...
movf   accumul,w    ; // G=0x30,...,Z=0x43
iorwf  accumuh,w    ; if ((char_io == 0) &&
btfss  STATUS,Z     ; (accumul == 0) && (accumuh == 0)) {
bra    monchr7      ; numbase &= ~2; // digit(s) leading 0(s),
bcf    numbase,1    ; char_io = 0;
clrf   char_io      ; break; // just go into octal mode
zos_RFI

monchr7
movlw  0xf0         ;
andwf  char_io,w    ;
btfss  STATUS,Z     ; } else if ((char_io & 0xf0 == 0) // 0-9,a-f
bra    monsave      ; && (numbase & 0x10)) { // base 16
btfss  numbase,4    ;
bra    monchr8      ;
swapf  accumuh,f    ;
movlw  0xf0         ;
andwf  accumuh,f    ; accumul <= 4;
swapf  accumul,w    ;
andlw  0x0f         ;
iorwf  accumuh,f    ; accumuh |= accumul >> 4;
movlw  0x0f         ;
andwf  char_io,f    ; char_io &= 0x0f;
andwf  accumul,f    ; accumul &= 0x0f;
swapf  accumul,w    ;
iorwf  char_io,w    ; accumul = (accumul << 4) | char_io;
movwf  accumul      ; char_io = 0;
clrf   char_io      ; break;
zos_RFI

```

```

monchr8
movf   char_io,w    ; } else /*if (char_io <= 9)*/ {
andlw  0xf0         ; uint16_t sum;
btfss  STATUS,Z     ; accumuh <= 1;
bra    monsave      ; accumuh |= (accumul & 0x80) ? 1 : 0;
                                ; accumul <= 1;
                                ; w = accumul; //w keeps original accumul<<1
                                ; accumuh <= 1;
                                ; accumuh |= (accumul & 0x80) ? 1 : 0;
                                ; accumul <= 1;
                                ; accumuh |= (accumul & 0x80) ? 1 : 0;
                                ; accumul <= 1; // accumuh:accumul <= 3;
                                ; if (numbase & 2) { // base 10 presumed
                                ; sum = (accumuh<<8)+accumul + w;
                                ; accumul = sum & 0x00ff;
                                ; accumuh = sum >> 8;
                                ; }
                                ; sum = (accumuh<<8)+accumul + char_io&0x0f;
                                ; accumul = sum & 0x00ff;
                                ; accumuh = sum >> 8;
                                ; break;
                                ; }
                                ; } // if we get here, restore input character
                                ; char_io += 0x37; // 0x10->'G',0x11->'H' etc.
                                ; zOS_AR1 = accumul;
zos_RFI

monchr9
movlw  0-0x37       ; if (isr) goto isr; // with zOS_AR1=accumul

monsave
movlw  0x37         ; } // switch ()
addwf  char_io,f    ; char_io = 0;
movf   accumul,w    ; } // if () // was less than 32 so aborts
movwf  zOS_AR1      ;
if (isr)
pagesel isr
goto   isr          ; zOS_RFI(); // reached only if isr == 0
else
zos_RFI
endif

;;;
monprmp
movf   1+destreg,w  ;monprmp:
movwf  accumuh      ; accumuh = destreg>>8;
iorwf  destreg,w    ; if (destreg) { // prompt with destreg if nonzero
pagesel monhex
btfsc  STATUS,Z     ; monhex(zos_job, p0);
bra    $+6          ; accumuh = destreg & 0xff;
call   monhex       ; monlsb(zos_job, p0);
movf   destreg,w    ; }
movwf  accumuh      ;monlast: zOS_ACC(&accumul,&numbase); zOS_RFI();
pagesel monlsb
call   monlsb       ; char_io = 0;
pagesel monspc
call   monspc       ; putchar(' ');

monzero
zos_ACC accumul,numbase

monlast
clrf   char_io      ;} // zOS_MON()
zos_RFI

endmon
zos_INP p,ra,rt,h,pi,monisr
endm

zos_NAM macro str
local start
start
dt     str
dt     0

```

```

dt      start-$
endm

zOS_MAN macro p,rat,rts,hb,pin,isr ;inline void zOS_MAN(int8_t p, int8_t rat,
pagesel endman
goto      endman          ;          int8_t* hb, int8_t pin) {

    local  mantask,manisr,manchr,manchr0,reenable,manchr1,manchr2,manchr3
    local  manchr4,manchr5,manchr6,manchr7,manchr8,manchr9,mannone,jobinfo
    local  manname,manloop,clrf,stkinfo,stkloop,endman

    local  p0,p1,wrap,t0scale,isradrl,isradrh,tskadrl,tskadrh,optadrl
    local  optadrh,accumul,accumuh,numbase,destreg,destreh,char_io,buf,max

    ;; 0x20~24 reserved for zOS_CON
p0      set    0x20
p1      set    0x21
wrap    set    0x22
t0scale set    0x23

    ;; 0x24~28 reserved for zOS_INP
isradrl set    0x24
isradrh set    0x25
tskadrl set    0x26
tskadrh set    0x27

    ;; 0x28~2F reserved for zOS_MON and derivations e.g. zOS_MAN
optadrl set    0x28
optadrh set    0x29
accumul set    0x2a
accumuh set    0x2b
numbase set    0x2c
destreg set    0x2d
destreh set    0x2e
char_io set    0x2f
buf      set    0x30
max      set    0x70

;copy the preceding lines rather than including this file, as definitions for
;zOS_MON()-derived macros referring to these local variables wouldn't open it
;until expansion and would throw an undefined-var error during the processing

mantask
#if 0;seems unnec 18 Jan
    movf    zOS_JOB,w      ;int8_t mantask(void) { //destreg,accumul,char_io
    movwf    BSR           ; bsr = zos_job; // to access char_io
#endif

    movf    char_io,w      ; if (char_io == 0)
    btfsc   STATUS,Z       ; return 0; // back to zOS_CON task
    return  ; switch (char_io) {

        xorlw 'G'         ;
        btfss STATUS,Z    ; caseG:
        bra  manchr       ; case 'G': // Generate a fork/duplicate of job
        clrf char_io      ; char_io = 0; // presume failure, so no retry

        movf    accumul,w  ; if (accumul == 0)
        btfsc   STATUS,Z   ; return 0;
        return  ; zOS_ARG(0, accumul);
        zOS_ARG 0
        zOS_ACC accumul,numbase
        movlw   'J'        ; zOS_ACC(&accumul, &numbase); // reset
        movwf   char_io    ; if (zOS_SWI(zOS_FRK))
        zOS_SWI zOS_FRK
        andlw   0x07       ; goto caseJ; // success, prints in job list
        btfsc   STATUS,Z   ; else
        clrf    char_io    ; break; // failure, drop to end of switch()

```

```

manchr
    movf    char_io,w      ;
    xorlw   'H'           ;
    btfss   STATUS,Z       ; caseH:
    bra     manchr0        ; case 'H': // find jobs by Handle (start addr)
    clrf    char_io       ; char_io = 0;

    movf    accumul,w      ; if (accumul == 0)
    iorwf   accumul,w      ;
    btfsc   STATUS,Z       ; return 0;
    return  ; zOS_ARG(0, accumul);
    movf    accumul,w      ;
    zOS_ARG 0
    movf    accumul,w      ;
    zOS_ARG 1
    zOS_ACC accumul,numbase
    movlw   'J'           ; zOS_ACC(&accumul, &numbase);
    movwf   char_io       ; if (zOS_SWI(zOS_FND))
    zOS_SWI zOS_FND
    andlw   0x07          ; goto caseJ; // FIXME: table, from match down
    movwf   accumul       ;
    btfsc   STATUS,Z       ; else
    clrf    char_io       ; break;

manchr0
    movf    char_io,w      ;
    xorlw   'I'           ;
    btfss   STATUS,Z       ; caseI:
    bra     manchr1        ; case 'I': // send a software Interrupt > 7
    clrf    char_io       ; char_io = 0; // with destreg zOS_AR1:zOS_AR0

    movf    destreg,w      ; zOS_ARG(0, destreg);
    clrf    destreg       ;
    zOS_ARG 0
    movf    1+destreg,w    ; zOS_ARG(1, destreh);
    clrf    1+destreg      ;
    zOS_ARG 1
    movf    accumul,w      ; w = accumul;
    zOS_ACC accumul,numbase
    andlw   0xf8           ; zOS_ACC(&accumul, &numbase); // reset
    btfsc   STATUS,Z       ; if (w & 0xf8) {
    bra     reenabl        ; int w = zOS_SWI(accumul); // disable again
    movlp   0              ; INTCON &= ~(1<<GIE); // for zOS_AR and _BUF()
    call    0x02           ; zOS_ARG(1, w);
    zOS_ARG 0              ; zOS_ARG(0, 0);

    #if 0
        clrf    zOS_AR1    ; zOS_BUF(zos_job, p0); // print hex SWI result
        xorwf   zOS_AR1,f  ; zOS_ENA();
        xorwf   zOS_AR0,f  ; goto caseJ;
        zOS_BUF FSR0,max,p0

    #else
        zOS_ARG 1          ;
        xorwf   zOS_AR0,f  ;
        zOS_SWI 0xff       ;
        movlw   '\r'       ;
        zOS_ARG 0          ;
        zOS_SWI 0xff       ;
        movlw   '\n'       ;
        zOS_ARG 0          ;
        zOS_SWI 0xff       ;

    #endif

    reenabl
    zOS_ENA

manchr1
    movf    char_io,w      ; }
    xorlw   'J'           ;
    btfss   STATUS,Z       ; caseJ:

```

```

bra    manchr2        ; case 'J': // List struct for all running jobs

decf    accumul,w      ; // keep char_io='J' until last job line prints
andlw   0x07           ;
btfsc   WREG,2         ; if ((accumul < 1) || (accumul > 5))
movlw   zOS_NUM-1      ;
addlw   0x01           ;
movwf   accumul        ; accumul = zOS_NUM;
bcf     INTCON,GIE     ; INTCON &= ~(1<<GIE); // to keep p0==p1 atomic
pagesel jobinfo
movf    p0,w           ;
xorwf   p1,w           ; if (p0 == p1)
btfsc   STATUS,Z       ; return jobinfo(); // will decrement accumul
goto    jobinfo        ; zOS_ENA(); // re-enable interrupts if p0!=p1
zOS_ENA
retlw   0              ; return 0; // try again after caller advances p0

manchr2
movf    char_io,w      ;
xorlw   'K'            ;
btfss   STATUS,Z       ; caseK:
bra     manchr3        ; case 'K': // Kill a single job (# mandatory)
clrf    char_io        ; char_io = 0;

movf    accumul,w      ; if (accumul == 0)
btfsc   STATUS,Z       ; return 0;
return  ; zOS_ARG(0, accumul);
zOS_ARG 0
zOS_ACC accumul,numbase
movlw   'J'            ; zOS_ACC(&accumul, &numbase);
movwf   char_io        ; zOS_SWI(zOS_END); // listed indicates failure
zOS_SWI zOS_END
;;; FIXME: put J at bottom so K onward don't pay a performance penalty awaiting

manchr3
movf    char_io,w      ;
xorlw   'L'            ;
btfss   STATUS,Z       ; caseL:
bra     manchr4        ; case 'L': // Launch a fresh instance of a job
clrf    char_io        ; char_io = 0;

movf    accumul,w      ; if (accumul == 0)
btfsc   STATUS,Z       ; return 0;
return  ; zOS_ARG(0, accumul);
zOS_ARG 0
zOS_ACC accumul,numbase
movlw   'J'            ; zOS_ACC(&accumul, &numbase); // reset
movwf   char_io        ; if ((w = zOS_SWI(zOS_FRK)) != 0) {
zOS_SWI zOS_FRK
andlw   0x07           ; zOS_ARG(0,w); zOS_SWI(zOS_RST);
btfsc   STATUS,Z       ; goto caseJ; // success, prints in job list
clrf    char_io        ; } else
zOS_ARG 0
zOS_SWI zOS_RST        ; break; // failure, drop to end of switch()

manchr4
movf    char_io,w      ;
xorlw   'N'            ;
btfss   STATUS,Z       ; caseN:
bra     manchr5        ; case 'N': // New (parameterless) job at addr

movf    accumul,w      ;
movwf   FSR0L          ;
movf    accumuh,w      ;
movwf   FSR0H          ;
clrw    ;
zOS_ARG 0
zOS_ARG 1
zOS_ARG 2

zOS_ARG 3
zOS_SWI zOS_NEW
zOS_ARG 0
zOS_BUF FSR0,max,p0
movlw   'J'           ;
movwf   char_io       ;

;
movf    accumul,w      ; if (accumul == 0)
btfsc   STATUS,Z       ; return 0;
return  ; zOS_ARG(0, accumul);
clrw    ;
zOS_ARG 0
zOS_ACC accumul,numbase
movlw   'J'           ; zOS_ACC(&accumul, &numbase);
movwf   char_io       ; if ((w = zOS_SWI(zOS_SLP)) != 0) {
zOS_SWI zOS_SLP
; andlw 0xff           ; accumul = w;
; movwf accumul        ; goto caseJ;
; btfsc STATUS,Z       ; } else
; clrf char_io         ; break;

manchr5
movf    char_io,w      ;
xorlw   'P'            ;
btfss   STATUS,Z       ; caseP:
bra     manchr6        ; case 'P': // Pause job by putting it to Sleep
clrf    char_io        ; char_io = 0;

movf    accumul,w      ; if (accumul == 0)
btfsc   STATUS,Z       ; return 0;
return  ; fsrl = 0x10 * (1 + accumul) + zOS_PCH;
movlw   'J'           ;
movwf   char_io       ;
zOS_MEM FSR1,accumul,zOS_PCH
movf    INDF1,w        ; if (*fsrl) { // is a valid (PCH not 0x00) job
btfsc   STATUS,Z       ; *fsr |= 0x80;
clrf    char_io        ; goto caseJ;
iorlw   0x80           ; } else {
movf    INDF1,f        ;
btfss   STATUS,Z       ;
movwf   INDF1          ; zOS_ACC(&accumul, &numbase);
btfsc   STATUS,Z       ; break; // only clear accumul if not caseJ
bra     manchr6        ; }
zOS_ACC accumul,numbase

manchr6
movf    char_io,w      ;
xorlw   'Q'            ;
btfss   STATUS,Z       ; caseQ:
bra     manchr7        ; case 'Q': // Quit without wake (off)
clrf    char_io        ; char_io = 0;

bcf     WDTCON,SWDTEN  ; WDTCON &= ~(1<<SWDTEN);
movf    accumul,f      ;
btfss   STATUS,Z       ; if (accumul)
sleep   ; sleep(); // never wakes up

manchr7
movf    char_io,w      ;
xorlw   'R'            ;
btfss   STATUS,Z       ; caseR:
bra     manchr8        ; case 'R': // Resume a pause/asleep job
clrf    char_io        ; char_io = 0;

swapf   accumul,w      ; if (accumul == 0x5a /*e.g.*/)
xorwf   accumul,w      ;
addlw   1              ;
btfsc   STATUS,Z       ;
reset   ; reset();

```

```

movf    accumul,w      ; if (accumul == 0)
btfsc   STATUS,Z       ; return 0;
return  ; fsr1 = 0x10 * (1 + accumul) + zOS_PCH;
movlw   'J'            ;
movwf   char_io        ; if (*fsr1 &= ~(1<<zOS_WAI)) {
zOS_MEM FSR1,accumul,zOS_PCH
movlw   0x7f           ; goto caseJ; // valid job won't be 0 or 0x80
andwf   INDF1,f        ; } else {
btfss   STATUS,Z       ; zOS_ACC(&accumul, &numbase);
bra     manchr8        ;
zOS_ACC accumul,numbase
clrf    char_io        ; break;

manchr8
movf    char_io,w      ; }
xorlw   'S'            ;
btfss   STATUS,Z       ;
bra     manchr9        ; case 'S': // Stack dump is actually scratch
clrf    char_io        ; char_io = 0; // always succeeds, no arg

decf    accumul,w      ; // keep char_io='S' until last job line prints
andlw   0x07           ;
btfsc   WREG,2         ; if ((accumul < 1) || (accumul > 5))
movlw   zOS_NUM-1      ;
addlw   0x01           ;
movwf   accumul        ; accumul = zOS_NUM;
bcf     INTCON,GIE     ; INTCON &= ~(1<<GIE); // to keep p0==p1 atomic
pagesel stkinf        ;
movf    p0,w           ;
xorwf   p1,w           ; if (p0 == p1)
btfsc   STATUS,Z       ; return jobinfo(); // will decrement accumul
goto    stkinf         ; zOS_ENA(); // re-enable interrupts if p0!=p1
zOS_ENA
retlw   0              ; return 0; // try again after caller advances p0

manchr9
movf    char_io,w      ;
xorlw   'Z'            ;
btfss   STATUS,Z       ;
bra     mannone        ; case 'Z': // go to low-power Zz mode for time
clrf    char_io        ; char_io = 0;

bsf     WDTCON,SWDTEN  ; if (w = accumul<<1) { // WDT prescaler
lslf    accumul,w      ; w |= 1<<SWDTEN; // enable the wakeup
btfsc   STATUS,Z       ;
bra     mannone        ;
iorlw   1<<SWDTEN      ;
movwf   WDTCON         ;
sleep   ; break; // wakes up according to prescaler

mannone
retlw   0              ; } return 0; // naught to do }

;guaranteed to arrive with p0=p1, interrupts off and in the correct bank

stkinf
movf     wrap,f        ;int8_t stkinf(void) {
movwf    p0            ; p0 = p1 = wrap;
movwf    p1            ;
movlw    low zOS_STK   ;
movwf    FSR0L         ;
movlw    high zOS_STK  ;
movwf    FSR0H         ;
decf     accumul,w     ;
brw      ;
addfsr   FSR0,6        ;
addfsr   FSR0,6        ;
addfsr   FSR0,6        ;
addfsr   FSR0,6        ; fsr0 = zOS_STK + 6 * (5 - accumul);

```

```

zOS_LOC FSR1,zOS_JOB,buf
movlw   '\r'           ; fsr1 = (zOS_JOB << 7) + buf;
movwi   FSR1++         ;
movlw   '\n'           ;
movwi   FSR1++         ;
movlw   '-'            ;
movwi   FSR1++         ;
movf    accumul,w      ;
addlw   -12            ; // print this stack offset as -0/-1/-2/-3/-4
zOS_HEX
movwi   FSR1++         ; p1 += sprintf(p1, "\r\n-%1X", accumul & 7);
movlw   3              ;
movwf   accumuh        ; for (accumuh = 3; accumuh; accumuh--) {

stkloop
movlw   ' '            ;
movwi   FSR1++         ; p1 += sprintf(p1, " %04X", *((int*) fsr0));
moviw   --FSR0         ;
movwi   FSR1++         ;
moviw   --FSR0         ;
movwi   FSR1++         ;
decfsz  accumuh,f      ;
bra     stkloop        ; }

movf     FSR1L,w       ;
movwf    p1            ; w = accumul--; // return with w as nonzero job
movf     accumul,w     ; if (accumul == 0)
decf     accumul,f     ; char_io = 0; // final row in table was printed
btfsc    STATUS,Z      ; zOS_ENA(); // interrupts back ON!
clrf     char_io       ; return w;
zOS_ENA
return   ;} // stkinf()

;guaranteed to arrive with p0=p1, interrupts off and in the correct bank

jobinfo
movf     wrap,w        ;int8_t jobinfo(void) {
movwf    p0            ; p0 = p1 = wrap;
movwf    p1            ; fsr0 = 0x10 * (1 + accumul); //FIXME: 2+
zOS_MEM FSR0,accumul,0
zOS_LOC FSR1,zOS_JOB,buf
movlw   '\r'           ; fsr1 = (zOS_JOB << 7) + buf;
movwi   FSR1++         ;
movlw   '\n'           ;
movwi   FSR1++         ;
movf     accumul,w     ; // print this job number 5/4/3/2/1
zOS_HEX
movwi   FSR1++         ; p1 += sprintf(p1, "\r\n%1X", accumul);

moviw    zOS_HDH[FSR0] ;
andlw    1<<zOS_PRB    ;
movlw    '*'           ; // print '*' if the job is privileged else ':'
btfss    STATUS,Z      ;
movlw    '*'           ; p1 += sprintf(p1, "%c", (zOS_HDH[fsr0] &
movwi    FSR1++         ; (1<<zOS_PRB)) ? '*' : ':');

zOS_IHF zOS_HDH,FSR0,FSR1
zOS_IHF zOS_HDL,FSR0,FSR1
movlw   ' '            ;
movwi   FSR1++         ;
movlw   'P'           ; // print the 4-hex-digit header then PC
movwi   FSR1++         ;
movlw   'C'           ; p1 += sprintf(p1, "%04X PC",
movwi   FSR1++         ; (zOS_HDH[fsr0] << 8) + zOS_HDL[fsr0]);

moviw    zOS_PCH[FSR0] ;
andlw    1<<zOS_WAI    ;
movlw   '='           ; // print '=' if the job is sleeping else 'z'
btfss    STATUS,Z      ;
movlw   'z'           ; p1 += sprintf(p1, "%c", (zOS_PCH[fsr0] &
movwi    FSR1++         ; (1<<zOS_WAI)) ? 'z' : ':');

```

```

zos_IHF zOS_PCH,FSR0,FSR1
moviw zOS_PCH[FSR0] ; // drop out after PCH if 0 (job is deleted)
btfsc STATUS,Z ; pl += sprintf(pl, "%02X", zOS_PCH[fsr0]);
bra manname ; if (zOS_PCH[fsr0] & 0xff00) {
zos_IHF zOS_PCL,FSR0,FSR1
movlw ' ' ; // print the low byte of program counter
movwi FSR1++ ; pl += sprintf(pl, "%02X", zOS_PCL[fsr0]);
moviw zOS_ISH[FSR0] ;
btfsc STATUS,Z ; // drop out after PCL if no interrupt routine
bra manname ; if (zOS_ISH[fsr0] & 0xff00) {
movlw 'I' ;
movwi FSR1++ ;
movlw 'S' ;
movwi FSR1++ ;
movlw 'R' ;
movwi FSR1++ ;
movlw '@' ;
movwi FSR1++ ; // print ISR@ then 4-hex-digit routine addr
zos_IHF zOS_ISH,FSR0,FSR1
zos_IHF zOS_ISR,FSR0,FSR1
movlw '(' ; pl += sprintf(pl, " ISR@%04X",
movwi FSR1++ ; (zOS_ISH[fsr0] << 8) + zOS_ISR[fsr0]);
movlw 'h' ;
movwi FSR1++ ;
zos_IHF zOS_HIM,FSR0,FSR1
movlw 's' ;
movwi FSR1++ ; // print (hw HwIMask sw SwIMask) scrunched up
zos_IHF zOS_SIM,FSR0,FSR1
movlw ')' ; pl += sprintf(pl, "(h%02Xs%02X) ",
movwi FSR1++ ; zOS_HIM[fsr0], zOS_SIM[fsr0]);
manname
movlw ' ' ;
movwi FSR1++ ;
movlw 0x22 ;''' ;
movwi FSR1++ ;
moviw zOS_PCH[FSR0] ;
btfss STATUS,Z ;
bra manlive ; if (zOS_PCH[fsr0] == 0) {
movlw low mandead ; static char mandead = "<not running>";
movwf FSR0L ;
movlw high mandead ;
movwf FSR0H ; fsr0 = mandead;
movlw mandead-manlive ;
movwf char_io ; char_io = strlen(mandead);
bra manloop ;
mandead
zos_NAM "<not running>"
manlive
moviw zOS_HDL[FSR0] ; } else {
movwf char_io ;
moviw zOS_HDH[FSR0] ;
iorlw 0x80 ;
movwf FSR0H ; fsr0 = 0x8000 | (zOS_HDH[fsr0] << 8) ;
movf char_io,w ;
movwf FSR0L ; fsr0 |= zOS_HDL[fsr0];
movi --FSR0 ;
iorlw 0xe0 ;
movwf char_io ; char_io = 0xe0 | *--fsr0; // max 32? chars
#if 1
addwf FSR0L,f ;
btfss STATUS,C ;
decf FSR0H,f ; for (fsr0 -= char_io; ++char_io; fsr1++) {
#else
local manbit0,manbit1
movf FSR0L,w ;
addwf char_io,w ;
btfss WREG,7 ;
bra manbit0 ;

```

```

btfss FSR0L,7 ;
decf FSR0H,f ;
bra manbit1 ;
manbit0
btfsc FSR0L,7 ;
decf FSR0H,f ;
manbit1
movwf FSR0L ; for (fsr0 -= char_io; ++char_io; fsr1++) {
#endif
manloop
moviw FSR0++ ; char w = *fsr0++ ;
btfsc WREG,7 ;
bra crlf ; if ((w > '\0177') ||
addlw 0-0x20 ;
btfsc WREG,7 ;
bra crlf ; (w < ' '))
addlw 0x20 ; break;
movwi FSR1++ ; *fsr1 = w; // added to buffer
incfsz char_io,f ;
bra manloop ; }
crlf
movlw 0x22 ;''' ;
movwi FSR1++ ;
movlw '\r' ; }
movwi FSR1++ ; }
movlw '\n' ; // print a second \r\n, double-spacing table
movwi FSR1++ ; pl += sprintf(pl, "\r\n");
movlw 'J' ;
movwf char_io ;
movf FSR1L,w ;
movwf pl ; w = accumul--; // return with w as nonzero job
movf accumul,w ; if (accumul == 0)
decf accumul,f ; char_io = 0; // final row in table was printed
btfsc STATUS,Z ; zOS_ENA(); // interrupts back ON!
clrf char_io ; return w;
zos_ENA
return ;
endman
local vars,manl,manh
vars set 0x20
manl set optadrl-vars
manh set optadrh-vars
zos_MON p,rat,rts,hb,pin,isr
movlw low mantask ; zOS_MON(p,ra,rt,h,pi,manisr); //fsr0=swi,1=adr
movwi manl[FSR1] ; optadrl = mantask & 0x00ff;
movlw high mantask ; optadrh = mantask >> 8;
movwi manh[FSR1] ; } // zOS_MAN()
endm
;;; zOS_CLC is an extension of the zOS_MAN() job manager shell into an rpn calc-
;;; ulator, as an example of how to use and customize the above console macros
;;;
;;; Note: because the max call depth of zOS_MON's ISR is nonzero (1), the max
;;; call depth for jobs in a system invoking these macros is reduced from 3 to 2
;;;
;;; (job 0)
;;; zOS_CLC is invoked with an optional isr routine (for any custom extensions):
;;; First a jump over the clcistr code ends the macro expansion
;;; zOS_MAN is invoked with all the zOS_CON arguments and its clcistr address:
;;; zOS_MON is invoked with all the zOS_CON arguments (and the clcistr address)
;;; First a jump over zOS_MON's monisr and all its support functions (no task)
;;; zOS_INP is invoked with all the zOS_CON arguments (and monisr's address)
;;; Immediately a near branch to rxdecl over the rxtask and rxisr code:
;;; When run, rxtask first calls any code at nonzero optadrh:optadrl address
;;; then jumps to the mandatorily nonzero tskadrh:tskadrl task of zOS_CON
;;; When handling an interrupt, rxisr either handles a received character or
;;; jumps to the mandatorily nonzero isradrh:isradrl isr address of zOS_CON

```

```

;;; and if a received character the ISR in this case jumps to nonzero monisr
;;; Unlike most declarations, rxdecl not only declares but launches, tweaks:
;;; zOS_CON is invoked with the port,rate,rttsflag,heartbeat,pin arguments:
;;; Immediately a near branch to decl over the task and isr code:
;;; When run, task initializes the global pair, circular buffer and greets
;;; (if the pair was still zero) then cedes the core awaiting a character
;;; which it then sends and loops back (to the zOS_INP task, not its own!)
;;; When handling an interrupt, isr handles the heartbeat and Timer0 stuff
;;; (if hardware) else assumes that a software interrupt is a char to send
;;; since any other applicable situation was handled by rxisr pre-jump
;;; end of zOS_CON expansion
;;; zOS_LAU then immediately assigns a job bank to the zOS_CON instance and
;;; uses FSR1 to set locals isradrh:isradrl,tskadrh:tskadrl,optadrh:optadrl
;;; to values zOS_CON just put in zOS_ARG1:zOS_ARG0, FSR0 (left at latter)
;;; at which point it overwrites the Program Counter and Handle fields with
;;; rxtask, ISR field with rxisr and RX HWI mask using FSR0 (left at SWI)
;;; end of zOS_INP expansion
;;; FSR1 (pointing to optadrh:optadrl) then gets the address of the ensuing
;;; mantask code (no ISR) which is then jumped over
;;; end of zOS_MON expansion
;;; end of zOS_MAN expansion
;;; end of zOS_CLC expansion
;;; (job 0)
;;; Since the end of zOS_INP, FSR0 has been pointing to the job information byte
;;; for the SWI mask that the job is to listen on for characters to output, so
;;; movwi 0[FSR0] with w set to the appropriate value: 8, 16, 32, 64 or 128

zOS_CLC macro p,ra,rt,h,pi,isr;inline void zOS_CLC(int8_t p, int8_t ra, int8_t
    local endclc,clcisr,clcprmp,endclc

    pagesel endclc
    goto endclc ; rt, int8_t* h, int8_t pi, void(*isr)() {

    local p0,p1,wrap,t0scale,isradrl,isradrh,tskadrl,tskadrh,optadrl
    local optadrh,accumul,accumuh,numbase,destreg,destreh,char_io,buf,max

    ;; 0x20~24 reserved for zOS_CON
p0 set 0x20
p1 set 0x21
wrap set 0x22
t0scale set 0x23

    ;; 0x24~28 reserved for zOS_INP
isradrl set 0x24
isradrh set 0x25
tskadrl set 0x26
tskadrh set 0x27

    ;; 0x28~2F reserved for zOS_MON and derivations e.g. zOS_MAN
optadrl set 0x28
optadrh set 0x29
accumul set 0x2a
accumuh set 0x2b
numbase set 0x2c
destreg set 0x2d
destreh set 0x2e
char_io set 0x2f
buf set 0x30
max set 0x70

;copy the preceding lines rather than including this file, as definitions for
;zOS_MON()-derived macros referring to these local variables wouldn't open it
;until expansion and would throw an undefined-var error during the processing

```

```

    local clctbl;clcsize ; throws "Duplicate label or redefining symbol"
clcisr
    movf zOS_AR0,w ; switch (char_io = zOS_AR0) {
    zOS_T63 ;
clctbl

```

```

retlw ' '
retlw '!'
retlw 0x22
retlw '#'
retlw '$'
retlw '%'
retlw '&'
retlw '"'
retlw '('
retlw ')'
retlw '*' ;0 ;zos_mac() not defined for '*'
retlw '+'
retlw ','
retlw '-'
retlw '.'
retlw '/' ;0 ;zos_div() not defined for '/'
retlw '0'
retlw '1'
retlw '2'
retlw '3'
retlw '4'
retlw '5'
retlw '6'
retlw '7'
retlw '8'
retlw '9'
retlw ':'
retlw 0x3b
retlw '<'
retlw '='
retlw '>'
retlw '?'
retlw '@'
retlw 'A'
retlw 'B'
retlw 'C'
retlw 'D'
retlw 'E'
retlw 'F'
retlw 'G'
retlw 'H'
retlw 'I'
retlw 'J'
retlw 'K'
retlw 'L'
retlw 'M'
retlw 'N'
retlw 'O'
retlw 'P'
retlw 'Q'
retlw 'R'
retlw 'S'
retlw 'T'
retlw 'U'
retlw 'V'
retlw 'W'
retlw 'X'
retlw 'Y'
retlw 'Z'
retlw '[' ; '{' ;
retlw '\\ ; '|' ;
retlw ']' ; '}' ;
retlw '^' ; '~' ;
clcsize equ $-clctbl
if clcsize-0x3f
    error "bad size: ASCII translation table expected to span 0x20 to 0x5e"
endif
movwf char_io ;
xorlw '+' ;

```

```

        btfss    STATUS,Z      ;
        bra     clcchr2        ; case '+': // 16-bit signed/unsigned add

        movf     accumul,w      ;
        addwf    destreg,f      ;
        movf     accumul,w      ;
        addwfc   1+destreg,f    ; destreg += (accumul < 8) | accumul;
        bra     clcprmp        ; break;

clcchr2
        movf     char_io,w      ;
        xorlw    '-'            ;
        btfss    STATUS,Z      ;
        bra     clcchr3        ; case '-': // 16-bit signed/unsigned subtract

        movf     accumul,w      ;
        subwf    destreg,f      ;
        movf     accumul,w      ;
        subwfb   1+destreg,f    ; destreg -= (accumul < 8) | accumul;
        bra     clcprmp        ; break;

clcchr3
        movf     char_io,w      ;
        xorlw    '*'            ;
        btfss    STATUS,Z      ;
        bra     clcchr4        ; case '*': // 8-bit by 8-bit unsigned multiply

#ifdef zos_mac
        clrf     zOS_AR0        ; // invoker of macro must implement zos_mac():
        clrf     zOS_AR1        ; // input arg zOS_AR1:zOS_AR0 (accumulator)
        movf     accumul,w      ; // zOS_AR2 (factor 1)
        movwf    zOS_AR2        ; // zOS_AR3 (factor 2)
        movf     destreg,w      ; // output arg zOS_AR1:zOS_AR0 (product)
        movwf    zOS_AR3        ; zOS_AR0 = (uint16_t) 0;
                                ; zOS_AR2 = accumul & 0x00ff;

        zOS_LOC FSR0,zOS_JOB,char_io
        pagesel  zos_mac
        call     zos_mac        ; zOS_AR3 = destreg & 0x00ff;
        movf     zOS_AR0,w      ; fsr0 = &char_io; // temp register (as INDF0)
        movwf    destreg        ; zos_mac(&zOS_AR0 /* += */,
        movf     zOS_AR1,w      ; &zOS_AR2 /* * */, &zOS_AR3, fsr0);
        movwf    1+destreg      ; destreg = (uint16_t) zOS_AR0;
#elseif
        bra     clcprmp        ; break;
#endif

clcchr4
        movf     char_io,w      ;
        xorlw    '/'            ;
        btfss    STATUS,Z      ;
        bra     clcchr5        ; case '/': // 15-bit by 8-bit unsigned divide

#ifdef zos_div
        movf     destreg,w      ; // invoker of macro must implement zos_div():
        movwf    zOS_AR0        ; // input arg zOS_AR1:zOS_AR0 (dividend)
        movf     1+destreg,w    ; // zOS_AR2 (divisor)
        andlw    0x7f          ; // output arg zOS_AR1:zOS_AR0 (quotient/exc)
        movwf    zOS_AR1        ; zOS_AR0 = (uint16_t) destreg & 0x7fff;
        movf     accumul,w      ; zOS_AR2 = accumul & 0xff;
        movwf    zOS_AR2        ; fsr0 = &char_io; // temp register (as INDF0)

        zOS_LOC FSR0,zOS_JOB,char_io
        pagesel  zos_div
        call     zos_div        ; zos_div(&zOS_AR0 /* /= */,
        movf     zOS_AR0,w      ; &zOS_AR2, &zOS_AR3/*scratch*/, fsr0);
        movwf    destreg        ;
        movf     zOS_AR1,w      ;
        movwf    1+destreg      ; destreg = (uint16_t) zOS_AR0;
#elseif
        bra     clcprmp        ; break;
#endif

clcchr5
        movf     char_io,w      ;

```

```

        xorlw    '^'            ;
        btfss    STATUS,Z      ;
        bra     clcchr6        ; case '^': // 8-bit by 8-bit exponentiation

#ifdef zos_mac
        movlw    0x01          ; // invoker of macro must implement zos_mac():
        clrf     zOS_AR1        ; // input arg zOS_AR1:zOS_AR0 (accumulator)
        movf     accumul,f      ; // zOS_AR2 (factor 1)
        btfsc    STATUS,Z      ; // zOS_AR3 (factor 2)
        bra     clcexpl        ; // output arg zOS_AR1:zOS_AR0 (product)
#elseif
        clcexpl
        clrf     zOS_AR0        ; zOS_AR1 = 0;
        clrf     zOS_AR1        ; for (uint8_t w = 1; accumul > 0; accumul--) {
        movwf    zOS_AR2        ; zOS_AR0 = (uint16_t) 0;
        movf     destreg,w      ; zOS_AR2 = w;
        movwf    zOS_AR3        ; zOS_AR3 = destreg & 0x00ff;

        zOS_LOC FSR0,zOS_JOB,char_io
        pagesel  zos_mac
        call     zos_mac        ; fsr0 = &char_io; // temp register (as INDF0)
        movf     zOS_AR0,w      ; zos_mac(&zOS_AR0 /* += */,
        decfsz   accumul,f      ; &zOS_AR2 /* * */, &zOS_AR3, fsr0);
        bra     clcexpl        ; w = zOS_AR0;
#elseif
        clcexpl
        movwf    destreg        ; }
        clrf     1+destreg      ; destreg = ((uint16_t) zOS_AR1) << 8 | w;
#endif

        bra     clcprmp        ; break;

clcchr6
        movf     char_io,w      ;
        xorlw    '!'            ;
        btfss    STATUS,Z      ;
        bra     clcchr7        ; case '!': // 3-bit factorial

#ifdef zos_mac
        movlw    0x01          ; // invoker of macro must implement zos_mac():
        clrf     zOS_AR1        ; // input arg zOS_AR1:zOS_AR0 (accumulator)
        movf     accumul,f      ; // zOS_AR2 (factor 1)
        btfsc    STATUS,Z      ; // zOS_AR3 (factor 2)
        bra     clcexpl        ; // output arg zOS_AR1:zOS_AR0 (product)
        decfsz   accumul,f      ;
        bra     clcexpl        ;
#elseif
        clcfac0
        clrf     zOS_AR0        ; zOS_AR1 = 0;
        clrf     zOS_AR1        ; for (uint8_t w = 1; accumul-- > 1; accumul--) {
        movwf    zOS_AR2        ; zOS_AR0 = (uint16_t) 0;
        movf     destreg,w      ; zOS_AR2 = w;
        decf     destreg,f      ; zOS_AR3 = destreg-- & 0x00ff;
        movwf    zOS_AR3        ; fsr0 = &char_io; // temp register (as INDF0)

        zOS_LOC FSR0,zOS_JOB,char_io
        pagesel  zos_mac
        call     zos_mac        ; zos_mac(&zOS_AR0 /* += */,
        movf     zOS_AR0,w      ; &zOS_AR2 /* * */, &zOS_AR3, fsr0);
        decfsz   accumul,f      ; w = zOS_AR0;
        bra     clcexpl        ; }
#elseif
        clcfac1
        movwf    destreg        ; destreg = ((uint16_t) zOS_AR1) << 8 | w;
        clrf     1+destreg      ; // 1 <= destreg <= 720
#endif

        bra     clcprmp        ; break;

clcchr7
        movf     accumul,w      ; default: zOS_AR1 = accumul; if (isr) goto isr;
        movwf    zOS_AR1        ; } // caller may use zOS_AR1 or accumul:accumul
        pagesel  isr
        if(isr)
            goto  isr          ; zOS_RFI();
        else
            zOS_RFI
        endif

        clcprmp

```



```

        movlw    '\r'        ;
        pagesel monbufs
        call     monbufs     ;
        movlw    '\n'        ;
        pagesel monbufs
        call     monbufs     ;clcrmp:
        movf     1+destreg,w  ; moncrlf(zos_job, p0);
        movwf    accumuh     ; accumuh = destreg>>8; monhex(zos_job, p0);
        pagesel monhex
        call     monhex      ; accumuh = destreg & 0xff; monlsb(zos_job, p0);
        movf     destreg,w   ; moncrlf(zos_job, p0);
        movwf    accumuh     ;clclast:
        pagesel monlsb
        call     monlsb      ; zOS_ACC(&accumul,&numbase); zOS_RFI();
        movlw    '\r'        ;
        pagesel monbufs
        call     monbufs     ;
        movlw    '\n'        ;
        pagesel monbufs
        call     monbufs     ; char_io = 0;
        zOS_ACC accumul,numbase
clclast
        clrf     char_io     ;} // zOS_CLC()
        zOS_RFI
endclc
        zOS_MAN p,ra,rt,h,pi,clcisz
        endm

zOS_T63 macro
        local   chrtran
        addlw   0-0x1f        ;#define zOS_T63(w) \
        btfsc   WREG,7        ;\
        clrw    ;\
        andlw   0x3f          ;\
        pagesel chrtran       ;\
        call    chrtran       ; w = table[(w >= ' ') ? (w & 0x3f) : 0];\
        bra     $+0x42        ; /*must be followed by 63-char retlw string:*/\
chrtran
        brw     ; static char table[64] = "\0\
        retlw   0             ;/* zOS_T63() */
        endm

```